
Ash Framework
Welcome! Here you will find everything you need to know to get started with and use Ash. This documentation is best viewed on hexdocs.

  
    
  
  Dive In


	What is Ash?
	Get Started
	See the roadmap


  
    
  
  About the Documentation


Tutorials walk you through a series of steps to accomplish a goal. These are learning-oriented, and are a great place for beginners to start.

Topics provide a high level overview of a specific concept or feature. These are understanding-oriented, and are perfect for discovering design patterns, features, and tools related to a given topic.

How-to guides are goal-oriented recipes for accomplishing specific tasks. These are also good to browse to get an idea of how Ash works and what is possible with it.

Reference documentation is produced automatically from our source code. It comes in the form of module documentation and DSL documentation. This documentation is information-oriented. Use the sidebar and the search bar to find relevant reference information.

  Packages


The Ash ecosystem consists of numerous packages, all of which have their own documentation. If you can't find something in this documentation, don't forget to search in any potentially relevant package.

  
    
  
  Data Layers


	AshPostgres | PostgreSQL data layer

	AshSqlite | SQLite data layer

	AshCsv | CSV data layer

  Web

  Resource Utilities

	AshArchival | Archive resources instead of deleting them

	AshPaperTrail | Keep a history of changes to resources

	AshAdmin | A push-button admin interface

  Learn with Livebook


We have a basic step by step tutorial in Livebook that introduces you to Ash. No prior Ash knowledge is required.
The Livebook tutorial is self contained and separate from the documentation below.

  Goals


In this guide we will:
	Create a new Elixir application and add Ash as a dependency
	Create a simple set of resources and see how they can be used
	Go over some core concepts of Ash
	Find out what material might be good to visit next


  
    
  
  Requirements


If you want to follow along yourself, you will need the following things:
	Elixir and Erlang installed
	A text editor to make the changes that we make
	A terminal to run the examples using iex


  
    
  
  Steps


For this tutorial, we'll use examples based around creating a help desk.
We will make the following resources:
	Helpdesk.Support.Ticket
	Helpdesk.Support.Representative

The actions we will be able to take on these resources include:
	Opening a new Ticket
	Closing a Ticket
	Assigning a Ticket to a representative


  
    
  
  Create a new project


We first create a new project with the --sup flag to add a supervision tree. This will be necessary for other follow-up tutorials.
# In your terminal
mix new --sup helpdesk && cd helpdesk

It is a good idea to make it a git repository and commit the initial project. You'll be able to see what changes we made, and can save your changes once we're done.
# Run in your terminal
git init
git add -A
git commit -m "first commit"
git branch -M main

Open the project in your text editor, and we'll get started.

  
    
  
  Add Ash to your application


Add the ash and picosat_elixir dependencies to your mix.exs
defp deps do
  [
    {:ash, "~> 3.0"},
    {:picosat_elixir, "~> 0.2"}
  ]
end
And then run mix deps.get && mix deps.compile to install the dependencies

  
    
  
  Picosat installation issues?


If you have trouble compiling picosat_elixir, then replace {:picosat_elixir, "~> 0.2"} with {:simple_sat, "~> 0.1"} to use a simpler (but mildly slower) solver. You can always switch back to picosat_elixir later once you're done with the tutorial.


  
    
  
  Formatting


To ensure that your code stays formatted like the examples here, you can add :ash as an import dependency in your .formatter.exs:
[
  # ...
  import_deps: [..., :ash],
  # ...
]
Note
For more auto-formatting options, see the Development Utilities guide.

And run mix deps.get, to install the dependency.

  
    
  
  Building your first Ash Domain


The basic building blocks of an Ash application are Ash resources. They are tied together by a domain module, which will allow you to interact with those resources.

  
    
  
  Creating our first resource


Let's start by creating our first resource along with our first domain. We will create the following files:
	The domain Helpdesk.Support, in lib/helpdesk/support.ex
	Our Ticket resource Helpdesk.Support.Ticket, in lib/helpdesk/support/ticket.ex.

To create the required folders and files, you can use the following command in your terminal:
mkdir -p lib/helpdesk/support && touch $_/ticket.ex
touch lib/helpdesk/support.ex

Your project structure should now include the following files:
lib/
├─ helpdesk/
│  ├─ support/
│  │  ├─ ticket.ex
│  ├─ support.ex
Add the following to the files we created
# lib/helpdesk/support.ex

defmodule Helpdesk.Support do
  use Ash.Domain

  resources do
    resource Helpdesk.Support.Ticket
  end
end
# lib/helpdesk/support/ticket.ex

defmodule Helpdesk.Support.Ticket do
  # This turns this module into a resource
  use Ash.Resource, domain: Helpdesk.Support

  actions do
    # Use the default implementation of the :read action
    defaults [:read]

    # and a create action, which we'll customize later
    create :create
  end

  # Attributes are the simple pieces of data that exist on your resource
  attributes do
    # Add an autogenerated UUID primary key called `:id`.
    uuid_primary_key :id

    # Add a string type attribute called `:subject`
    attribute :subject, :string
  end
end
Next, add your domain to your config.exs
Run the following to create your config.exs if it doesn't already exist
mkdir -p config
touch config/config.exs
and add the following contents to it.
# in config/config.exs
import Config

config :helpdesk, :ash_domains, [Helpdesk.Support]

  
    
  
  Try our first resource out


Run iex -S mix in your project's root directory and try out the following.
To create a ticket, we first make an Ash.Changeset for the :create action of the Helpdesk.Support.Ticket resource. Then we pass it to the Ash.create!/1 function.
Helpdesk.Support.Ticket
|> Ash.Changeset.for_create(:create)
|> Ash.create!()
This returns what we call a record which is an instance of a resource.
#Helpdesk.Support.Ticket<
  ...,
  id: "c0f8dc32-a018-4eb4-8656-d5810118f4ea",
  subject: nil,
  ...
>

  
    
  
  Customizing our Actions


One thing you may have noticed earlier is that we created a ticket without providing any input, and as a result our ticket had a subject of nil. Additionally, we don't have any other data on the ticket. Lets add a status attribute, ensure that subject can't be nil, and provide a better interface by giving the :create action a better name, and accepting :subject as part of the action.
We'll start with the attribute changes:
# lib/helpdesk/support/ticket.ex

attributes do
  ...
  attribute :subject, :string do
    # Don't allow `nil` values
    allow_nil? false

    # Allow this attribute to be public. By default, all attributes are private.
    public? true
  end

  # status is either `open` or `closed`. We can add more statuses later
  attribute :status, :atom do
    # Constraints allow you to provide extra rules for the value.
    # The available constraints depend on the type
    # See the documentation for each type to know what constraints are available
    # Since atoms are generally only used when we know all of the values
    # it provides a `one_of` constraint, that only allows those values
    constraints [one_of: [:open, :closed]]

    # The status defaulting to open makes sense
    default :open

    # We also don't want status to ever be `nil`
    allow_nil? false
  end
end
And then replace the :create action with :open, and accept :subject as input.
# lib/helpdesk/support/ticket.ex

actions do
  ...
  create :open do
    accept [:subject]
  end
end
Let's try these changes in iex:
We use create! with an exclamation point here because that will raise the error which gives a nicer view of the error in iex
# Use this to pick up changes you've made to your code, or restart your session
recompile()

Helpdesk.Support.Ticket
|> Ash.Changeset.for_create(:open, %{subject: "My mouse won't click!"})
|> Ash.create!()
And we can see our newly created ticket with a subject and a status.
#Helpdesk.Support.Ticket<
  ...
  id: "3c94d310-7b5e-41f0-9104-5b193b831a5d",
  status: :open,
  subject: "My mouse won't click!",
  ...
>
If we didn't include a subject, or left off the arguments completely, we would see an error instead
** (Ash.Error.Invalid) Invalid Error

* attribute subject is required

  
    
  
  Updates and validations


Now let's add some logic to close a ticket. This time we'll add an update action.
Here we will use a change. Changes allow you to customize how an action executes with very fine-grained control. There are built-in changes that are automatically available as functions, but you can define your own and pass it in as shown below. You can add multiple, and they will be run in order. See the Ash.Changeset module documentation for more.
# lib/helpdesk/support/ticket.ex

actions do
  ...
  update :close do
    # We don't want to accept any input here
    accept []

    validate attribute_does_not_equal(:status, :closed) do
      message "Ticket is already closed"
    end

    change set_attribute(:status, :closed)
    # A custom change could be added like so:
    #
    # change MyCustomChange
    # change {MyCustomChange, opt: :val}
  end
end
Try out opening and closing a ticket in iex:
# Use this to pick up changes you've made to your code, or restart your session
recompile()

# parenthesis so you can paste into iex
ticket = (
  Helpdesk.Support.Ticket
  |> Ash.Changeset.for_create(:open, %{subject: "My mouse won't click!"})
  |> Ash.create!()
)

ticket
|> Ash.Changeset.for_update(:close)
|> Ash.update!()

#Helpdesk.Support.Ticket<
  ...
  status: :closed,
  subject: "My mouse won't click!",
  ...
>

  
    
  
  Querying without persistence


So far we haven't used a data layer that does any persistence, like storing records in a database. All that this simple resource does is return the record back to us. You can see this lack of persistence by attempting to use a read action:
Ash.read!(Helpdesk.Support.Ticket)
Which will raise an error explaining that there is no data to be read for that resource.
In order to save our data somewhere, we need to add a data layer to our resources. Before we do that, however, let's go over how Ash allows us to work against many different data layers (or even no data layer at all).
Resources without a data layer will implicitly be using Ash.DataLayer.Simple, which will just return structs and won't actually store anything. The way that we make our queries return some data is by leveraging context, a free-form map available on queries and changesets. The simple data layer looks for query.context[:data_layer][:data][resource]. It provides a utility, Ash.DataLayer.Simple.set_data/2 to set it.
Try the following in iex. We will open some tickets, and close some of them, and then use Ash.DataLayer.Simple.set_data/2 to use those tickets.
# Ash.Query is a macro, so it must be required
require Ash.Query

tickets =
  for i <- 0..5 do
    ticket =
      Helpdesk.Support.Ticket
      |> Ash.Changeset.for_create(:open, %{subject: "Issue #{i}"})
      |> Ash.create!()

    if rem(i, 2) == 0 do
      ticket
      |> Ash.Changeset.for_update(:close)
      |> Ash.update!()
    else
      ticket
    end
  end
Find the tickets where the subject contains "2". Note that the we're setting the ticket data that we're querying using set_data.
Helpdesk.Support.Ticket
|> Ash.Query.filter(contains(subject, "2"))
|> Ash.DataLayer.Simple.set_data(tickets)
|> Ash.read!()
Find the tickets that are closed and their subject does not contain "4"
Helpdesk.Support.Ticket
|> Ash.Query.filter(status == :closed and not(contains(subject, "4")))
|> Ash.DataLayer.Simple.set_data(tickets)
|> Ash.read!()
The examples above could be easily implemented with Enum.filter, but the real power here is to allow you to use the same tools when working with any data layer.
Even though it doesn't persist data in any way, Ash.DataLayer.Simple can be useful to model static data, or be used for resources where all the actions are manual and inject data from other sources.

  
    
  
  Adding basic persistence


Before we get into working with relationships, let's add some real persistence to our resource. This will let us add relationships and try out querying data.
There is a built in data layer that is useful for testing and prototyping, that uses ETS. ETS (Erlang Term Storage) is OTP's in-memory database, so the data won't actually stick around beyond the lifespan of your program, but it's a simple way to try things out.
To add it to your resource, modify it like so:
# lib/helpdesk/support/ticket.ex

use Ash.Resource,
  domain: Helpdesk.Support,
  data_layer: Ash.DataLayer.Ets
Now we can slightly modify our code above, by removing the Ash.DataLayer.Simple.set_data/2 calls, and we can see our persistence in action. Remember, ETS is in-memory, meaning restarting your application/iex session will remove all of the data.
# Use this to pick up changes you've made to your code, or restart your session
recompile()

require Ash.Query

for i <- 0..5 do
  ticket =
    Helpdesk.Support.Ticket
    |> Ash.Changeset.for_create(:open, %{subject: "Issue #{i}"})
    |> Ash.create!()

  if rem(i, 2) == 0 do
    ticket
    |> Ash.Changeset.for_update(:close)
    |> Ash.update!()
  end
end

# Show the tickets where the subject contains "2"
Helpdesk.Support.Ticket
|> Ash.Query.filter(contains(subject, "2"))
|> Ash.read!()

# Show the tickets that are closed and their subject does not contain "4"
Helpdesk.Support.Ticket
|> Ash.Query.filter(status == :closed and not(contains(subject, "4")))
|> Ash.read!()

  
    
  
  Adding relationships


Now we want to be able to assign a Ticket to a Representative. First, let's create the Representative resource:
# lib/helpdesk/support/representative.ex

defmodule Helpdesk.Support.Representative do
  # This turns this module into a resource using the in memory ETS data layer
  use Ash.Resource,
    domain: Helpdesk.Support,
    data_layer: Ash.DataLayer.Ets

  actions do
    # Add the default simple actions
    defaults [:read]

    create :create do
      accept [:name]
    end
  end

  # Attributes are the simple pieces of data that exist on your resource
  attributes do
    # Add an autogenerated UUID primary key called `:id`.
    uuid_primary_key :id

    # Add a string type attribute called `:name`
    attribute :name, :string do
      # Make the attribute public in order to give a name when calling functions from `Ash.Changeset`.
      public? true
    end
  end

  relationships do
    # `has_many` means that the destination attribute is not unique, therefore many related records could exist.
    # We assume that the destination attribute is `representative_id` based
    # on the module name of this resource and that the source attribute is `id`.
    has_many :tickets, Helpdesk.Support.Ticket
  end
end
Now let's modify our Ticket resource to have the inverse relationship to the Representative.
# lib/helpdesk/support/ticket.ex

relationships do
  # belongs_to means that the destination attribute is unique, meaning only one related record could exist.
  # We assume that the destination attribute is `representative_id` based
  # on the name of this relationship and that the source attribute is `representative_id`.
  # We create `representative_id` automatically.
  belongs_to :representative, Helpdesk.Support.Representative
end
Finally, let's add our new Representative resource to our domain module
# lib/helpdesk/support.ex

resources do
 ...
 resource Helpdesk.Support.Representative
end
You may notice that if you don't add the resource to your domain, or if you don't add the belongs_to relationship, that you'll get helpful errors at compile time. Helpful compile time validations are a core concept of Ash as we really want to ensure that your application is valid.

  
    
  
  Working with relationships


The simplest way to work with belongs to relationships is to allow directly editing the underlying id field.

  
    
  
  managing relationships


There are a wide array of options when managing relationships, and we won't cover all of them here. See the Managing Relationships guide for more.

Add the assign action to allow us to assign a Ticket to a Representative.
# lib/helpdesk/support/ticket.ex

update :assign do
  accept [:representative_id]
end
Let's try it out in our iex console!
Use recompile to pick up changes you've made to your code, or just restart your session.
recompile()

  
    
  
  Open a Ticket


ticket = (
  Helpdesk.Support.Ticket
  |> Ash.Changeset.for_create(:open, %{subject: "I can't find my hand!"})
  |> Ash.create!()
)

  
    
  
  Create a Representative


representative = (
  Helpdesk.Support.Representative
  |> Ash.Changeset.for_create(:create, %{name: "Joe Armstrong"})
  |> Ash.create!()
)

  
    
  
  Assign that Representative to the Ticket


ticket
|> Ash.Changeset.for_update(:assign, %{representative_id: representative.id})
|> Ash.update!()

  
    
  
  What next?


What you've seen above barely scratches the surface of what Ash can do. In a lot of ways, it will look very similar to other tools that you've seen. If all that you ever used was the above, then realistically you won't see much benefit to using Ash.
Where Ash shines however, is all of the tools that can work with your resources. You have the ability to extend the framework yourself, and apply consistent design patterns that enable unparalleled efficiency, power and flexibility as your application grows.
Get Help
	Check out ElixirForum
	Join our Discord server
	Open a GitHub issue

Persist your data
See The AshPostgres getting started guide to see how to back your resources with Postgres. This is highly recommended, as the Postgres data layer provides tons of advanced capabilities.
Add a web API
Check out AshJsonApi and AshGraphql extensions to build APIs around your resource
Authorize access and work with users
See the Policies guide for information on how to authorize access to your resources using actors and policies.
Clean up your code that uses Ash?
Creating and using changesets manually can be verbose, and they all look very similar. Luckily, Ash has your back and can help you build high quality interfaces for you!
Check out the Code Interface Guide to derive things like Helpdesk.Support.Ticket.assign!(representative.id)



  

    
What is Ash?
    

Ash is an opinionated, composable set of application building blocks designed for extensibility. It shines when building web apps, APIs and services (though it can be used for any kind of Elixir application). It has integrations for Phoenix, LiveView, Postgres, GraphQL, Oban, and many more.
To achieve this Ash provides a "Resource" abstraction you use to model the heart of your application. These resources then determine the database schema, API endpoints, state machines, background jobs, and more. Resources are the source of truth for your entire application, and everything stems from them.
Ash is a declarative framework with deep extensibility. We provide a suite of extensions informed from building production apps, as well as a toolkit so you can build your own. We provide escape hatches ranging from the simple and small to the ability to override large pieces of behavior. Your Ash application is just an Elixir application, so if you want to do something completely custom then Ash won’t get in your way.
It is not a web framework, like Phoenix or Rails. It is a framework for building your application layer, independent of how it is exposed or consumed. It is not an alternative to frameworks like Phoenix, rather a complement to them.
Model your domain, derive the rest
Ash derives significant portions of your application directly from your resources, with little to no effort required. This allows you to focus on what matters most: your business logic.
We leverage the best of the Elixir ecosystem under the hood, providing a single unified tool-chain for our users.


A good analogy is design systems in the world of web development. When you use a design system, you get a set of components that are designed to work together, and you can build your application by combining these components in different ways. Ash is like a design system for your application's domain model.
Or, for a less technical analogy, Ash is like a fully stocked workshop. When you arrive at the workshop, you may need to learn where everything is, but once you do, you have everything you need to build anything you can dream up.

  
    
  
  Why should I use it?


The fundamental idea behind Ash is that when the various components of your system can have consistent expectations of how the other components around them work, you can ultimately do a significant amount more, with less. For example, once you've defined your resources, it takes only a few additional lines of code to have your database structure generated, and a full featured API built around it.
Ash has many use cases, with varying degrees of complexity. Ash helps you on day 1 of your project, removing boiler plate and allowing you to focus on the essential complexity of your application. It also helps you on year 5, lending consistency, code reuse, and maintainability.
Ash is a force multiplier
Things that once took days or weeks can be done in hours, and to a degree of quality that would have been unreasonable before. Entire classes of bugs are eliminated. And the best part is, you can do all of this without sacrificing the flexibility, robustness and ecosystem that Elixir is known for.


  
    
  
  Watch the ElixirConf 2023 Talk





  

    
Design Principles
    

The design principles behind Ash allows us to build an extremely flexible and powerful set of tools, without locking users into specific choices at any level. The framework acts as a spinal cord for your application, with extension points at every level to allow for custom behavior. What follows are the core tenets behind Ash Framework.

  
    
  
  Anything, not Everything


"Anything, not Everything" means building a framework capable of doing anything, not providing a framework that already does everything. The first is possible, the second is not. Our primary goal is to provide a framework that unlocks potential, and frees developers to work on the things that make their application special.
To this end, there are many prebuilt extensions to use, but there is also a rich suite of tools to build your own extensions. In this way, you can make the framework work for you, instead of struggling to fit your application to a strictly prescribed pattern. Use as much of Ash as you can, and leverage the amazing Elixir ecosystem for everything else.

  
    
  
  Declarative, Introspectable, Derivable


The real superpower behind Ash is the declarative design pattern. All behavior is driven by explicit, static declarations. A resource, for example, is really just a configuration file. On its own it does nothing. It is provided to code that reads that configuration and acts accordingly.
You can read more about some simple declarative design patterns outside of the context of Ash Framework in An Incremental Approach to Declarative Design.

  
    
  
  Configuration over Convention


While convention has value, we believe that explicit configuration ultimately leads to more discoverable, maintainable and flexible applications than a convention driven approach. This means that we never do things like assume that files with a given name are a certain type of thing, or that because a file is in a certain location, it should perform a specific function.

  
    
  
  Pragmatism First


While Ash does have lofty goals and a roadmap, the priority for development is always what the current users of Ash need or are having trouble with. We focus on simple, pragmatic, and integrated solutions that meld well with the rest of the framework.
A high priority is placed on ensuring that our users don't experience feature whip-lash due to poorly thought out implementations, and that any breaking changes (a rare occurrence) have a clean and simple upgrade path. This is something made much easier by the declarative pattern.

  
    
  
  Community


The Ash community comes together and collaborates to make sure that we can all build our software quickly, effectively and in a way that will age gracefully. We have a strict code of conduct, and love working with people of any experience level or background. To experience this first-hand, participate on ElixirForum or join our discord!

  
    
  
  Domain Driven Design?


 Ash is not a Domain Driven Design framework, if we're talking about "proper" Domain Driven Design as it is taught and discussed today. Domain Driven Design comes with a considerable amount of baggage and unnecessary complexity. While we identify with the goals of Domain Driven Design, we believe that a simpler approach is more effective, and that much of what DDD teaches are actually implementation details, and not design concepts. If the name wasn't taken, we would surely have claimed it for ourselves. If you must have a similar term for Ash, consider it a "Resource-oriented, Declarative Design Application Framework".




  

    
Contributing to Ash
    


  
    
  
  Welcome!


We are delighted to have anyone contribute to Ash, regardless of their skill level or background. We welcome contributions both large and small, from typos and documentation improvements, to bug fixes and features. There is a place for everyone's contribution here. Check the issue tracker or join the ElixirForum/discord server to see how you can help! Make sure to read the rules below as well.

  
    
  
  Rules


	We have a zero tolerance policy for failure to abide by our code of conduct. It is very standard, but please make sure
you have read it.
	Issues may be opened to propose new ideas, to ask questions, or to file bugs.
	Before working on a feature, please talk to the core team/the rest of the community via a proposal. We are
building something that needs to be cohesive and well thought out across all use cases. Our top priority is
supporting real life use cases like yours, but we have to make sure that we do that in a sustainable way. The
best compromise there is to make sure that discussions are centered around the use case for a feature, rather
than the proposed feature itself.
	Before starting work, please comment on the issue and/or ask in the discord if anyone is handling an issue. Be aware that if you've commented on an issue that you'd like to tackle it, but no one can reach you and/or demand/need arises sooner, it may still need to be done before you have a chance to finish. However, we will make all efforts to allow you to finish anything you claim.




  

    
Alternatives
    

There aren't really any alternatives to Ash that we are aware of that do all of the same things, but there are many different packages out there that do some of the things that Ash does.
This is a living document, and is not comprehensive. We are not vouching for any of these packages, but rather listing them here for your convenience to investigate on your own.
Want to add or edit this list? Open a pull request Want a more comprehensive list? Check out the Awesome Elixir.

  
    
  
  Application Frameworks


These frameworks have similar overarching goals of helping you build your application layer.
	Commanded - An event sourced application framework.
	Sleeky - Billed as a lightweight alternative to Ash. Inspired by Ash, but more tightly built on top of ecto.


  
    
  
  Application Design


	Boundary - A library for defining boundaries in your application.
	Phoenix Contexts - Phoenix ships with a concept called "contexts", which provide some generators with application design guidance.


  
    
  
  Building APIs


	Absinthe - A GraphQL toolkit for Elixir. This is what AshGraphql uses under the hood, but you can also use Absinthe directly.
	Phoenix - Phoenix is a web framework for Elixir. It is not necessarily an API framework, but has all the tools you need to build APIs by hand.
	JSONAPI Elixir - A library for building JSONAPI compliant APIs in Elixir on top of Ecto.
	Open API Spex - A library for generating OpenAPI (Swagger) documentation for your API. We generate these for you in AshJsonApi, but you can use this library to build open api specifications of your hand-written API


  
    
  
  Working with Data


	Ecto - Ecto is a database wrapper and query generator for Elixir. In many cases, Ash uses Ecto under the hood, but it is also available to build on top of directly.
	Flop - A library designed to easily apply filtering, ordering, and pagination to Ecto queries.


  
    
  
  Authentication


	mix phx.gen.auth - A mix task that generates authentication for a Phoenix application. Some folks prefer to use this over AshAuthentication even if they are using Ash.
	Assent - Multi-provider authentication framework.


  
    
  
  Authorization


	Bodyguard - A phoenix-context-based policy authorization framework.


  
    
  
  Validation


	Ecto - Ecto can be used to validate data at the edge, using things like schema-less changesets.
	Drops - a collection of small modules that provide useful extensions and functions that can be used to work with data effectively.
	GuardedStruct - validation, sanitization, and construction of structs, supporting nested structs.




  

    
Changelog
    


  
    
  
  v3.0.6 (2024-05-23)



  
    
  
  Bug Fixes:


	improve non-atomic-upgrade policy handling

	ensure we have a primary key for certain kinds of calculations:

	handle subquery-requiring calculations in calculate/2

	mixup between atomic upsert vs update capability of datalayers (#1198)



  
    
  
  v3.0.5 (2024-05-23)



  
    
  
  Bug Fixes:


	only use Mix.deps_tree if defined


  
    
  
  Improvements:


	add atomic_upgrade? flag to update/destroy actions

	do not do atomic upgrade by default unless require_atomic? is true

	allow configuring the read action used by atomic upgrades



  
    
  
  v3.0.4 (2024-05-22)



  
    
  
  Features:


	Ash.Reactor: Add bulk_update step type. (#1185)

	allow retrieving the count of paginated relationships (#1183)

	Ash.Reactor: Add bulk_create step type.



  
    
  
  Bug Fixes:


	ensure that all notifications are sent for bulk destroy/update

	properly hydrate and scope sorts

	handle list of atomic conditions (#1194) (#1195)

	handle nil value for old_values (#1191)

	allow all strategies for bulk actions in code interfaces by default

	use Ash.Sort.parse_input/3 in Ash.Query.sort_input/2

	improve Ash.Domain.Info typespecs

	make Ash.Resource.Validation.Changing work correctly in atomics (#1178)

	check the where condition before checking validation atomicity (#1177)

	handle nil notification results better

	don't emit after batch notifications if notify?: false

	prefer resource domain over option domain (#1176)

	don't require domain for empty stream bulk update and destroy (#1175)

	only return valid non nil values items from generator (#1121)

	properly validate input action on bulk destroy

	allow setting authorize?: false

	honor get? for bulk update/bulk destroy



  
    
  
  Improvements:


	support anonymous aggregates and calculations in queries

	Implement show_sensitive? config (#1180)

	support filter_input and sort_input in Ash.Query.build/2

	add template_requires_actor check for changesets

	don't use queries for streaming if they have hooks

	consider from: nil in changing_attributes/1 check



  
    
  
  3.0.3 (2024-05-22)



  
    
  
  Features:


	Ash.Reactor: Add bulk_update step type. (#1185)

	allow retrieving the count of paginated relationships (#1183)

	Ash.Reactor: Add bulk_create step type.



  
    
  
  Bug Fixes:


	properly hydrate and scope sorts

	handle list of atomic conditions (#1194) (#1195)

	handle nil value for old_values (#1191)

	allow all strategies for bulk actions in code interfaces by default

	use Ash.Sort.parse_input/3 in Ash.Query.sort_input/2

	improve Ash.Domain.Info typespecs

	make Ash.Resource.Validation.Changing work correctly in atomics (#1178)

	check the where condition before checking validation atomicity (#1177)

	handle nil notification results better

	don't emit after batch notifications if notify?: false

	prefer resource domain over option domain (#1176)

	don't require domain for empty stream bulk update and destroy (#1175)

	only return valid non nil values items from generator (#1121)

	properly validate input action on bulk destroy

	allow setting authorize?: false

	honor get? for bulk update/bulk destroy



  
    
  
  Improvements:


	support anonymous aggregates and calculations in queries

	Implement show_sensitive? config (#1180)

	support filter_input and sort_input in Ash.Query.build/2

	add template_requires_actor check for changesets

	don't use queries for streaming if they have hooks

	consider from: nil in changing_attributes/1 check



  
    
  
  v3.0.2 (2024-05-15)



  
    
  
  Improvements:


	[Ash.Expr] add pattern matching for clarity on values accepted by ref/1 and ref/2

	[Ash.Expr] add can_return_nil?/1 callback to Ash expressions, allowing for various optimizations

	[Ash.Type.NewType] raise argument error on unknown options in Ash.Type.NewType. Helps with typos & misunderstandings

	[embedded resources] use the source configuration for attributes in embedded resources (it was previously just ignored)

	[Ash.Policy.Authorizer] better type specification for checks, to get better autocomplete and compile time validation

	[Ash.Error.Invalid.NoSuchInput] added a did_you_mean field and used it in the error message



  
    
  
  Bug Fixes:


	[Ash.Resource] properly persist simple_notifiers (they were being ignored before)

	[code interface] accept single ids in code interface as subject for destroy/update

	[bulk update] ensure that the changed? context is set in after action hooks on batches

	[relationships] allow for inferred domains when authorizing join queries

	[Ash.Expr] don't treat nil as not a valid value when type casting lists

	[atomic upgrade] keep data's metadata in atomic upgraded update (#1165)



  
    
  
  v3.0.1 (2024-05-14)



  
    
  
  Features:


	[Ash.Resource.Change.Builtins] Add cascade_destroy to builtin changes.


  
    
  
  Bug Fixes:


	[calculations] calculation eager evaluation bug caused exists to eager evaluate when we didn't actually have the related data

	[field policies] fix field policy rewrite errors on non-success cases (#1163)

	[embedded resources] fix embedded resource authorization (#1159) (#1160)

	infinite recursion if query is empty (#1158)

	[Ash.DataLayer.Ets] ensure that changeset filters are honored in ETS destroy/update_query

	[update/destroy actions] don't rollback transactions on stale records, ignore stale records in bulk actions

	[bulk creates] don't check required belongs to fields until after setting them in bulk creation

	[code interface] check require_reference? when generating update code interface (#1152)



  
    
  
  v3.0.0



  
    
  
  3.0


We are starting the changelog fresh. See documentation/2.0-CHANGELOG.md in GitHub for the old changelogs.

  
    
  
  Breaking Changes:


For a guide on adjusting to these breaking changes, see the upgrade guide
	[Ash.Api] has been renamed to Ash.Domain, and references to the concept have been renamed as well, i.e in options and in the DSL
	[Ash] we now call functions on this, isntead of the domain. i.e Ash.create and Ash.read. The generated functions are now marked as deprecated
	[Ash] remove process context functionality. You can no longer store the actor/tenant in the context with Ash.set_actor and so on
	[private?] deprecate private?: false in favor of the more explicit public?: true
	[default_accept] default default_accept is now []
	[action lifecycle] after transaction hooks cannot be added from inside of other lifecycle hooks
	[Ash.NotLoaded] use %Ash.NotLoaded{} for unselected values, instead of nil
	[require_atomic?] now defaults to true, requiring opt-out of atomic behavior
	[authorization] default api.authorization.authorize to :by_default
	[Ash.Registry] has been removed
	[actions] domain must always be known when constructing changesets
	[Ash.Notifier] requires_original_data?/2 callback defaults to false
	[Ash.Notifier.PubSub] default to previous_values?: false, allowing notifications to be sent for atomic updates
	[unknown inputs] all action invocations now use UnknownInput errors when given an input they don't accept
	[policies] requires_original_data?/2 callback on checks defaults to false
	[Ash.Calculation] has been renamed to Ash.Resource.Calculation
	[Ash.Resource.Calculation] "strict mode" has been added and defaults to true. This causes only explicitly requested fields from relationships to be loaded
	[Ash.Query.Calculation] positional arguments are now an options list
	[calculations] anonymous function calculations in a resource now take lists and return lists, instead of a single record (like standard calculations do)
	[context] The context argument passed to many different callbacks is now a struct, tailored to that specific context. For example, in a calculation you will receive an Ash.Resource.Calculation.Context
	[after_action/before_action] These builtin changes now accept a 3rd context argument
	[picosat_elixir] is now optional (simple_sat is now an alternative)
	[Ash.Changeset] Ash.Changeset.new! has been removed
	[Ash.Changeset] Ash.Changeset.new/2 has been removed (Ash.Changeset.new/1 is still available)
	[Ash.Changeset] changeset.filters is now changeset.filter
	[Ash.Changeset] reverse order of before action & before transaction hooks. They now run in the action they are added. They used to run in reverse order.
	[Ash.CiString] Ash.CiString.new/1 returns nil on nil input
	[belongs_to.attribute_writable?] add attribute_public? for controlling publicity, and default attribute_writable? to true.
	[Ash.Filter.TemplateHelpers] removed, all functions needed for expressions are now defined in Ash.Expr
	[Ash.Expr] keyword lists are no longer special cased in ash expressions, and requiring pinning like any other value.
	[Ash.Resource] default read actions are now paginatable with keyset and offset pagination (but pagination is not required)
	[Ash.Resource] default actions require explicit accept lists (or will use default_accept). i.e defaults [:read, create: [:first_name, :last_name]]
	[Ash.Resource] simple_notifiers is now an option to use Ash.Resource, instead of being in the DSL at resource.simple_notifiers
	[Ash.Flow] has been removed and put in its own package ash_flow. It is being deprecated in favor of Reactor
	[Ash.Error] the implementation has been extracted out to Splode. Defining new Ash.Errors is now done by defining a new Splode.Error
	[Ash.Query] swap position of sort order and arguments in calculation sorting, i.e instead of calculation: {:asc, %{...args}} it is now calculation: {%{...args}, :asc}
	[Ash.Resource.Aggregate] add include_nil? aggregate option, and default it to false (so list and first aggregates do not consider nil values by default)
	[Ash.Policy.FilterCheck] now accepts context arguments, like Ash.Policy.FilterCheckWithContext
	[Ash.Policy.FilterCheckWithContext] has been removed, use Ash.Policy.FilterCheck


  
    
  
  Features:


	[Ash.Type] add new remove_nil_items? array type constraint (#1116)
	[Ash.Query] Paginatable relationships (#1050)
	[Ash.DataLayer] new calculate/3 callback that allows for data layers to compute the result of expressions outside the context of a query. Used to power Ash.calculate/3.
	[validations] new builtin validations, attributes_present/2 and attributes_absent/2
	[multitenancy] configurable multitenancy behaviour on read actions (#1030)
	[Ash.Reactor] Add new change step type which can be used to modify changesets.
	[Ash.Changeset] add Ash.Changeset.update_change/2 function and builtin change (#976)
	[Ash.Domain] code interfaces can now be defined on the domain
	[Ash.Domain] policies can now be defined on the domain, and will run before resource policies
	[Ash.ToTenant] add Ash.ToTenant, allowing for passing arbitrary values as tenants
	[Ash] add Ash.read_first (like Ash.read_one, but applies a limit automatically)
	[Ash] support a second optional input option for create, update and destroy, allowing for things like Ash.create!(Post, %{text: "text"}, opts)
	[sensitive?] support sensitive? option in query aggregate/calculation (#963)
	[Ash.Resource] support require_reference?: false on code interfaces, for when an update or destroy action uniquely identifies a record (or for bulk update/destroy)
	[Ash.Resource] notifiers can now be specified for specific actions, using the notifiers option
	[mix ash.rollback] delegates to extensions to trigger their rollback tasks
	[Ash.Query] add Ash.Query.apply_to/3, to "apply" the query to a set of records (i.e filter, sort, distinct, etc.)
	[Ash.CustomExpression] Use Ash.CustomExpression to extend Ash's expression syntax in a data-layer agnostic way
	[code interface] Code interface functions now support bulk actions, in a "do what I mean" way. For example: Domain.deactive(post) can also be Post |> Ash.Query.filter(active == true) |> Domain.deactive()


  
    
  
  Improvements:


	[Ash.Actions.Sort] allow providing a stream of records to sort, and performance improvements
	[bulk actions] add read_action option to bulk actions (#1088)
	[Ash.stream] support streaming with offset, or even no pagination
	[Ash.DataLayer.Ets] add debug logging, similar to ecto query debug logging
	[Ash.DataLayer.Ets] support update_query, destroy_query and Ash.Changeset.filter/2
	[Embedded resources] don't add autogenerated_id to embeds if they don't have a primary key
	[Ash.Resource] you can now omit the return type of generic actions, indicating it either succeeds or fails, returning :ok or {:error, error}
	[Ash.Resource] Generic actions can now accept a Reactor module, running it directly. (#993)
	[Ash.Resource] support allow_nil_input dsl option in update/destroy actions (#964)
	[Ash.Resource] The filter option can now be supplied multiple times in read actions and in relationships. They will be combined with and
	[Ash.Resource] private attributes can now be accepted as action inputs
	[Ash.Expr] is now imported automatically into places you will likely use it, like changes, validations, checks and calculations.
	[Ash.Query] is now required automatically in places you will likely use it, as above
	[sortable?] fields may mark themselves as unusable in sorts by using sortable? false
	[sensitive?] calculations and aggregates may now also be marked as sensitive?


  
    
  
  Bug Fixes:


	[Ash.Type] apply array type nil_items? constraint after item constraints are applied (#1115)
	[Ash.DataLayer.Ets] fix ETS data layer's support for lateral joining
	[bulk actions] ensure transaction is rolled back on data layer errors during streaming
	[bulk actions] set notify?: true when return_notifications?: true is set
	[Ash.Changeset] attributes_present?/2 -> attribute_present?/2
	[Ash.Filter] don't eager evaluate type/3 because data layers require type information
	[Ash.Changeset] when comparing identities for manage_relationship, we now properly cast the values. Before, "1" and 1 were not considered equal for integer primary keys/identity fields
	Many more bug fixes were added, but few are relevant enough to list here




  

    
Domains
    

Domains serve three primary purposes:
	They group related resources together, providing organization and structure to your project.
	They allow you to define a centralized code interface
	They allow you to configure certain cross-cutting concerns of those resources in a single place.

If you are familiar with a Phoenix Context, you can think of a domain as the Ash equivalent.

  
    
  
  Grouping Resources


In an Ash.Domain, you will typically see something like this:
defmodule MyApp.Tweets do
  use Ash.Domain

  resources do
    resource MyApp.Tweets.Tweet
    resource MyApp.Tweets.Comment
  end
end
With this definition, you can do things like placing all of these resources into a GraphQL Api with AshGraphql. You'd see a line like this:
use AshGraphql, domains: [MyApp.Tweets]

  
    
  
  Centralized Code Interface


Working with our domain & resources in code can be done the long form way, by building changesets/queries/action inputs and calling the relevant function in Ash. However, we generally want to expose a well defined code API for working with our resources. This makes our code much clearer, and gives us nice things like auto complete and inline documentation.
defmodule MyApp.Tweets do
  use Ash.Domain

  resources do
    resource MyApp.Tweets.Tweet do
      # define a function called `tweet` that uses
      # the `:create` action on MyApp.Tweets.Tweet
      define :tweet, action: :create, args: [:text]
    end

    resource MyApp.Tweets.Comment do
      # define a function called `comment` that uses
      # the `:create` action on MyApp.Tweets.Comment
      define :comment, action: :create, args: [:tweet_id, :text]
    end
  end
end
With these definitions, we can now do things like this:
tweet = MyApp.Tweets.tweet!("My first tweet!", actor: user1)
comment = MyApp.Tweets.comment!(tweet.id, "What a cool tweet!", actor: user2)

  
    
  
  Configuring Cross-cutting Concerns



  
    
  
  Built in configuration


Ash.Domain comes with a number of built-in configuration options. See d:Ash.Domain for more.
For example:
defmodule MyApp.Tweets do
  use Ash.Domain

  resources do
    resource MyApp.Tweets.Tweet
    resource MyApp.Tweets.Comment
  end

  execution do
    # raise the default timeout for all actions in this domain from 30s to 60s
    timeout :timer.seconds(60)
  end

  authorization do
    # disable using the authorize?: false flag when calling actions
    authorize :always
  end
end

  
    
  
  Extensions


Extensions will often come with "domain extensions" to allow you to configure the behavior of all resources within a domain, as it pertains to that extension. For example:
defmodule MyApp.Tweets do
  use Ash.Domain,
    extensions: [AshGraphql.Domain]

  graphql do
    # skip authorization for these resources
    authorize? false
  end

  resources do
    resource MyApp.Tweets.Tweet
    resource MyApp.Tweets.Comment
  end
end

  
    
  
  Policies


You can also use Ash.Policy.Authorizer on your domains. This allows you to add policies that apply to all actions using this domain. For example:
defmodule MyApp.Tweets do
  use Ash.Domain,
    extensions: [Ash.Policy.Authorizer]

  resources do
    resource MyApp.Tweets.Tweet
    resource MyApp.Tweets.Comment
  end

  policies do
    # add a bypass up front to allow administrators to do whatever they want
    bypass actor_attribute_equals(:is_admin, true) do
      authorize_if always()
    end

    # forbid all access from disabled users
    policy actor_attribute_equals(:disabled, true) do
      forbid_if always()
    end
  end
end



  

    
Attributes
    

Attributes specify the name, type and additional configuration of a simple property of a record. When using SQL data layers, for example, an attribute would correspond to a column in a database table. For information on types, see Ash.Type.
To see all of the options available when building attributes, see d:Ash.Resource.Dsl.attributes.attribute
If you are looking to compute values on demand, see the Calculations guide and the aggregates guide.

  
    
  
  Special attributes


In Ash there are 4 special attributes these are:
	create_timestamp
	update_timestamp
	integer_primary_key
	uuid_primary_key

These are really just shorthand for an attribute with specific options set. They're outlined below.

  
    
  
  create_timestamp


You may recognise this if you have used Ecto before. This attribute will record the time at which each row is created, by default it uses DateTime.utc_now/1.
create_timestamp :inserted_at is equivalent to an attribute with these options:
attribute :inserted_at, :utc_datetime_usec do
  writable? false
  default &DateTime.utc_now/0
  match_other_defaults? true
  allow_nil? false
end

  
    
  
  update_timestamp


This is also similar in Ecto. This attribute records the last time a row was updated, also using DateTime.utc_now/1 by default.
update_timestamp :updated_at is equivalent to:
attribute :updated_at, :utc_datetime_usec do
  writable? false
  default &DateTime.utc_now/0
  update_default &DateTime.utc_now/0
  match_other_defaults? true
  allow_nil? false
end

  
    
  
  uuid_primary_key


This attribute is used in almost every resource. It generates a UUID every time a new record is made.
uuid_primary_key :id is equivalent to:
attribute :id, :uuid do
  writable? false
  default &Ash.UUID.generate/0
  primary_key? true
  allow_nil? false
end

  
    
  
  integer_primary_key


Creates a generated integer primary key. Keep in mind that not all data layers support auto incrementing ids, but for SQL data layers this is a very common practice. For those that don't, it is your own job to provide values for the primary key. We generally suggest using UUIDs over integers, as there are a lot of good reasons to not use autoincrementing integer ids.
integer_primary_key :id is equivalent to:
attribute :id, :integer do
  writable? false
  generated? true
  primary_key? true
  allow_nil? false
end



  

    
Relationships
    

Relationships describe the connections between resources and are a core component of Ash. Defining relationships enables you to do things like
	Loading related data
	Filtering on related data
	Managing related records through changes on a single resource
	Authorizing based on the state of related data


  
    
  
  Relationships Basics


A relationship exists between a source resource and a destination resource. These are defined in the relationships block of the source resource. For example, if MyApp.Tweet is the source resource, and MyApp.User is the destination resource, we could define a relationship called :owner like this:
defmodule MyApp.Tweet do
  use Ash.Resource,
    data_layer: my_data_layer

  attributes do
    uuid_primary_key :id
    attribute :body, :string
  end

  relationships do
    belongs_to :owner, MyApp.User
  end
end

  
    
  
  Managing related data


See Managing Relationships for more information.
Your data layer may enforce foreign key constraints, see the following guides for more information:
	AshPostgres references


  
    
  
  Kinds of relationships


There are four kinds of relationships:
	belongs_to
	has_one
	has_many
	many_to_many

Each of these relationships has a source resource and a destination resource with a corresponding attribute on the source resource (source_attribute), and destination resource (destination_attribute). Relationships will validate that their configured attributes exist at compile time.
You don't need to have a corresponding "reverse" relationship for every relationship, i.e if you have a MyApp.Tweet resource with belongs_to :user, MyApp.User you aren't required to have a has_many :tweets, MyApp.Tweet on MyApp.User. All that is required is that the attributes used by the relationship exist.

  
    
  
  Belongs To


# on MyApp.Tweet
belongs_to :owner, MyApp.User
A belongs_to relationship means that there is an attribute (source_attribute) on the source resource that uniquely identifies a record with a matching attribute (destination_attribute) in the destination. In the example above, the source attribute on MyApp.Tweet is :owner_id and the destination attribute on MyApp.User is :id.
Attribute Defaults
By default, the source_attribute is defined as :<relationship_name>_id of the type :uuid on the source resource and the destination_attribute is assumed to be :id. You can override the attribute names by specifying the source_attribute and destination_attribute options like so:
belongs_to :owner, MyApp.User do
  # defaults to :<relationship_name>_id (i.e. :owner_id)
  source_attribute :custom_attribute_name

  # defaults to :id
  destination_attribute :custom_attribute_name
end
You can further customize the source_attribute using options such as:
	d:Ash.Resource.Dsl.relationships.belongs_to|define_attribute? to define it yourself
	d:Ash.Resource.Dsl.relationships.belongs_to|attribute_type to modify the default type
	d:Ash.Resource.Dsl.relationships.belongs_to|attribute_public? to make the source attribute public?: true

For example:
belongs_to :owner, MyApp.User do
  attribute_type :integer
  attribute_writable? false
end
Or if you wanted to define the attribute yourself,
attributes do
  attribute :owner_foo, MyApp.CustomType
end

...
relationships do
  belongs_to :owner, MyApp.User do
    define_attribute? false
    source_attribute :owner_foo
  end
end
Customizing default belongs_to attribute type
Destination attributes that are added by default are assumed to be :uuid. To change this, set the following configuration in config.exs:
config :ash, :default_belongs_to_type, :integer
See the docs for more: d:Ash.Resource.Dsl.relationships.belongs_to

  
    
  
  Has One


# on MyApp.User
has_one :profile, MyApp.Profile
A has_one relationship means that there is a unique attribute (destination_attribute) on the destination resource that identifies a record with a matching unique attribute (source_resource) in the source. In the example above, the source attribute on MyApp.User is :id and the destination attribute on MyApp.Profile is :user_id.
A has_one is similar to a belongs_to except the reference attribute is on
the destination resource, instead of the source.
Attribute Defaults
By default, the source_attribute is assumed to be :id, and destination_attribute defaults to <snake_cased_last_part_of_module_name>_id.
See the docs for more: d:Ash.Resource.Dsl.relationships.has_one

  
    
  
  Has Many


# on MyApp.User
has_many :tweets, MyApp.Tweet
A has_many relationship means that there is a non-unique attribute (destination_attribute) on the destination resource that identifies a record with a matching unique attribute (source_resource) in the source. In the example above, the source attribute on MyApp.User is :id and the destination attribute on MyApp.Tweet is :user_id.
A has_many relationship is similar to a has_one because the reference attribute exists on the destination resource. The only difference between this and has_one is that the destination attribute is not unique, and therefore will produce a list of related items. In the example above, :tweets corresponds to a list of MyApp.Tweet records.
Attribute Defaults
By default, the source_attribute is assumed to be :id, and destination_attribute defaults to <snake_cased_last_part_of_module_name>_id.
See the docs for more: d:Ash.Resource.Dsl.relationships.has_many

  
    
  
  Many To Many


A many_to_many relationship can be used to relate many source resources to many destination resources. To achieve this, the source_attribute and destination_attribute are defined on a join resource. A many_to_many relationship can be thought of as a combination of a has_many relationship on the source/destination resources and a belongs_to relationship on the join resource.
For example, consider two resources MyApp.Tweet and MyApp.Hashtag representing tweets and hashtags. We want to be able to associate a tweet with many hashtags, and a hashtag with many tweets. To do this, we could define the following many_to_many relationship:
# on MyApp.Tweet
many_to_many :hashtags, MyApp.Hashtag do
  through MyApp.TweetHashtag
  source_attribute_on_join_resource :tweet_id
  destination_attribute_on_join_resource :hashtag_id
end
The through option specifies the "join" resource that will be used to store the relationship. We need to define this resource as well:
defmodule MyApp.TweetHashtag do
  use Ash.Resource,
    data_layer: your_data_layer

  postgres do
    table "tweet_hashtags"
    repo MyApp.Repo
  end

  relationships do
    belongs_to :tweet, MyApp.Tweet, primary_key?: true, allow_nil?: false
    belongs_to :hashtag, MyApp.Hashtag, primary_key?: true, allow_nil?: false
  end

  actions do
    defaults [:read, :destroy, create: :*, update: :*]
  end
end
It is convention to name this resource <source_resource_name><destination_resource_name> however this is not required. The attributes on the join resource must match the source_attribute_on_join_resource and destination_attribute_on_join_resource options on the many_to_many relationship. The relationships on the join resource are standard belongs_to relationships, and can be configured as such. In this case, we have specified that the :tweet_id and :hashtag_id attributes form the primary key for the join resource, and that they cannot be nil.
Now that we have a resource with the proper attributes, Ash will use this automatically under the hood when
performing relationship operations like filtering and loading.
See the docs for more: d:Ash.Resource.Dsl.relationships.many_to_many

  
    
  
  Loading related data


There are two ways to load relationships:
	in the query using Ash.Query.load/2
	directly on records using Ash.load/3


  
    
  
  On records


Given a single record or a set of records, it is possible to load their relationships by calling the load function on the record's parent domain. For example:
# user = %User{...}
Ash.load(user, :tweets)

# users = [%User{...}, %User{...}, ....]
Ash.load(users, :tweets)
This will fetch the tweets for each user, and set them in the corresponding tweets key.
%User{
  ...
  tweets: [
    %Tweet{...},
    %Tweet{...},
    ...
  ]
}
See Ash.load/3 for more information.

  
    
  
  In the query


The following will return a list of users with their tweets loaded identically to the previous example:
User
|> Ash.Query.load(:tweets)
|> Ash.read()
At present, loading relationships in the query is fundamentally the same as loading on records. Eventually, data layers will be able to optimize these loads (potentially including them as joins in the main query).
See Ash.Query.load/2 for more information.

  
    
  
  More complex data loading


Multiple relationships can be loaded at once, i.e
Ash.load(users, [:tweets, :followers])
Nested relationships can be loaded:
Ash.load(users, followers: [:tweets, :followers])
The queries used for loading can be customized by providing a query as the value.
followers = Ash.Query.sort(User, follower_count: :asc)

Ash.load(users, followers: followers)
Nested loads will be included in the parent load.
followers =
  User
  |> Ash.Query.sort(follower_count: :asc)
  |> Ash.Query.load(:followers)

# Will load followers and followers of those followers
Ash.load(users, followers: followers)

  
    
  
  no_attributes? true


This is really useful when creating customized relationships that aren't joined with simple attribute matches. For example:
has_many :higher_priority_tickets, __MODULE__ do
  no_attributes? true
  # parent/1 in this case puts the expression on this current resource
  # so this is "tickets with priority higher than this ticket"
  filter expr(priority > parent(priority))
end
This can also be very useful when combined with multitenancy. Specifically, if you have a tenant resource like Organization,
you can use no_attributes? to do things like has_many :employees, Employee, no_attributes?: true, which lets you avoid having an
unnecessary organization_id field on Employee. The same works in reverse: has_one :organization, Organization, no_attributes?: true
allows relating the employee to their organization.

  
    
  
  Caveats for using no_attributes?


	You can still manage relationships from one to the other, but "relate" and "unrelate" will have no effect, because there are no fields to change.
	Loading the relationship on a list of resources will not behave as expected in all circumstances involving multitenancy. For example, if you get a list of Organization and then try to load employees, you would need to set a single tenant on the load query, meaning you'll get all organizations back with the set of employees from one tenant. This could eventually be solved, but for now it is considered an edge case.



  
    
  
  Manual Relationships


Manual relationships allow you to express complex or non-typical relationships between resources in a standard way. Individual data layers may interact with manual relationships in their own way, so see their corresponding guides. In general, you should try to use manual relationships sparingly, as you can do a lot with filters on relationships, and the no_attributes? flag.

  
    
  
  Example


In our Helpdesk example, we'd like to have a way to find tickets
In the Representative resource, define a has_many relationship as manual and point to the module where
it will be implemented.
relationships do
  has_many :tickets_above_threshold, Helpdesk.Support.Ticket do
    manual Helpdesk.Support.Ticket.Relationships.TicketsAboveThreshold
  end
end
Using Ash to get the destination records is ideal, so you can authorize access like normal
but if you need to use a raw ecto query here, you can. As long as you return the right structure.
The TicketsAboveThreshold module is implemented as follows.
defmodule Helpdesk.Support.Ticket.Relationships.TicketsAboveThreshold do
  use Ash.Resource.ManualRelationship
  require Ash.Query

  def load(records, _opts, %{query: query, actor: actor, authorize?: authorize?}) do
    # Use existing records to limit resultds
    rep_ids = Enum.map(records, & &1.id)

    {:ok,
     query
     |> Ash.Query.filter(representative_id in ^rep_ids)
     |> Ash.Query.filter(priority > representative.priority_threshold)
     |> Helpdesk.Support.read!(actor: actor, authorize?: authorize?)
     # Return the items grouped by the primary key of the source, i.e representative.id => [...tickets above threshold]
     |> Enum.group_by(& &1.representative_id)}
  end
end

  
    
  
  Reusing the Query


Since you likely want to support things like filtering your relationship when being loaded, you will want to make sure that you use the query being provided. However, depending on how you're loading the relationship, you may need to do things like fetch extra records. To do this, you might do things like
def load(records, _opts, %{query: query, ..}) do
  # unset some fields
  fetch_query = Ash.Query.unset(query, [:limit, :offset])

  # or, to be more safe/explicit, you might make a new query, explicitly setting only a few fields
  fetch_query = query.resource |> Ash.Query.filter(^query.filter) |> Ash.Query.sort(query.sort)

  ...
end

  
    
  
  Fetching the records and then applying a query


Lets say the records come from some totally unrelated source, or you can't just modify the query to fetch the records you need. You can fetch the records you need and then apply the query to them in memory.
def load(records, _opts, %{query: query, ..}) do
  # fetch the data from the other source, which is capabale of sorting
  data = get_other_data(data, query.sort)

  query
  # unset the sort since we already applied that
  |> Ash.Query.unset([:sort])
  # apply the query in memory (filtering, distinct, limit, offset)
  |> Ash.Query.apply_to(data)
end

  
    
  
  Managing Relationships


In Ash, managing related data is done via Ash.Changeset.manage_relationship/4. There are various ways to leverage the functionality expressed there. If you are working with changesets directly, you can call that function. However, if you want that logic to be portable (e.g available in ash_graphql mutations and ash_json_api actions), then you want to use the following argument + change pattern:
actions do
  update :update do
    argument :add_comment, :map do
      allow_nil? false
    end

    argument :tags, {:array, :uuid} do
      allow_nil? false
    end

    # First argument is the name of the action argument to use
    # Second argument is the relationship to be managed
    # Third argument is options. For more, see `Ash.Changeset.manage_relationship/4`. This accepts the same options.
    change manage_relationship(:add_comment, :comments, type: :create)

    # Second argument can be omitted, as the argument name is the same as the relationship
    change manage_relationship(:tags, type: :append_and_remove)
  end
end
With this, those arguments can be used in action input:
post
|> Ash.Changeset.for_update(:update, %{tags: [tag1.id, tag2.id], add_comment: %{text: "comment text"}})
|> Ash.update!()

  
    
  
  Argument Types


Notice how we provided a map as input to add_comment, and a list of UUIDs as an input to manage_relationship. When providing maps or lists of maps, you are generally just providing input that will eventually be passed into actions on the destination resource. However, you can also provide individual values or lists of values. By default, we assume that value maps to the primary key of the destination resource, but you can use the value_is_key option to modify that behavior. For example, if you wanted adding a comment to take a list of strings, you could say:
argument :add_comment, :string

...
change manage_relationship(:add_comment, :comments, type: :create, value_is_key: :text)
And then you could use it like so:
post
|> Ash.Changeset.for_update(:update, %{tags: [tag1.id, tag2.id], add_comment: "comment text"})
|> Ash.update!()

  
    
  
  Derived behavior


Determining what will happen when managing related data can be complicated, as the nature of the problem itself is quite complicated. In some simple cases, like type: :create, there may be only one action that will be called. But in order to support all of the various ways that related resources may need to be managed, Ash provides a rich set of options to determine what happens with the provided input. Tools like AshPhoenix.Form can look at your arguments that have a corresponding manage_relationship change, and derive the structure of those nested forms. Tools like AshGraphql can derive complex input objects to allow manipulating those relationships over a graphql Api. This all works because the options are, ultimately, quite explicit. It can be determined exactly what actions might be called, and therefore what input could be needed.
To see all of the options available, see Ash.Changeset.manage_relationship/4



  

    
Calculations
    

Calculations in Ash allow for displaying complex values as a top level value of a resource.

  
    
  
  Primer



  
    
  
  Declaring calculations on a resource



  
    
  
  Expression Calculations


The simplest kind of calculation refers to an Ash expression. For example:
calculations do
  calculate :full_name, :string, expr(first_name <> " " <> last_name)
end
See the Expressions guide for more.

  
    
  
  Module Calculations


When calculations require more complex code or can't be pushed down into the data layer, a module that uses Ash.Resource.Calculation can be used.
defmodule Concat do
  # An example concatenation calculation, that accepts the delimiter as an argument,
  #and the fields to concatenate as options
  use Ash.Resource.Calculation

  # Optional callback that verifies the passed in options (and optionally transforms them)
  @impl true
  def init(opts) do
    if opts[:keys] && is_list(opts[:keys]) && Enum.all?(opts[:keys], &is_atom/1) do
      {:ok, opts}
    else
      {:error, "Expected a `keys` option for which keys to concat"}
    end
  end

  @impl true
  # A callback to tell Ash what keys must be loaded/selected when running this calculation
  # you can include related data here, but be sure to include the attributes you need from said related data
  # i.e `posts: [:title, :body]`.
  def load(_query, opts, _context) do
    opts[:keys]
  end

  @impl true
  def calculate(records, opts, %{separator: separator}) do
    Enum.map(records, fn record ->
      Enum.map_join(opts[:keys], separator, fn key ->
        to_string(Map.get(record, key))
      end)
    end)
  end

  # You can implement this callback to make this calculation possible in the data layer
  # *and* in elixir. Ash expressions are already executable in Elixir or in the data layer, but this gives you fine grain control over how it is done
  # @impl true
  # def expression(opts, context) do
  # end
end

# Usage in a resource
calculations do
  calculate :full_name, :string, {Concat, keys: [:first_name, :last_name]} do
    # You need to use the [allow_empty?: true, trim?: false] constraints here.
    # The separator could be an empty string or require a leading or trailing space,
    # but would be trimmed or even set to `nil` without the constraints shown below.
    argument :separator, :string do
      allow_nil? false
      constraints [allow_empty?: true, trim?: false]
      default ""
    end
  end
end
See the documentation for the calculations section in Resource DSL docs and the Ash.Resource.Calculation docs for more information.
The calculations declared on a resource allow for declaring a set of named calculations that can be used by extensions.
They can also be loaded in the query using Ash.Query.load/2, or after the fact using Ash.load/3. Calculations declared on the resource will be keys in the resource's struct.

  
    
  
  Custom calculations in the query


Example:
User
|> Ash.Query.calculate(:full_name, {Concat, keys: [:first_name, :last_name]}, :string, %{separator: ","})
See the documentation for Ash.Query.calculate/4 for more information.

  
    
  
  Arguments in calculations


Using the above example with arguments, you can load a calculation with arguments like so:
load(full_name: [separator: ","])
If the calculation uses an expression, you can also filter and sort on it like so:
query
|> Ash.Query.filter(full_name(separator: " ") == "Zach Daniel")
|> Ash.Query.sort(full_name: {%{separator: " "}, :asc})



  

    
Aggregates
    

Aggregates in Ash allow for retrieving summary information over groups of related data. A simple example might be to show the "count of published posts for a user". Aggregates allow us quick and performant access to this data, in a way that supports being filtered/sorted on automatically. More aggregate types can be added, but you will be restricted to only the supported types. In cases where aggregates don't suffice, use Calculations, which are intended to be much more flexible.

  
    
  
  Declaring aggregates on a resource


Example:
aggregates do
  count :count_of_posts, :posts do
    filter expr(published == true)
  end
end
The available aggregate types are:
	count - counts related items meeting the criteria.
	exists - checks if any related items meet the criteria.
	first - gets the first related value matching the criteria. Must specify the field.
	sum - sums the related items meeting the criteria. Must specify the field.
	list - lists the related values. Must specify the field.
	max - gets the maximum related value. Must specify the field.
	min - gets the minimum related value. Must specify the field.
	avg - gets the average related value. Must specify the field.
	custom - allows for a custom aggregate. Implementation depends on the data layer. Must provide an implementation.

The declared set of named aggregates can be used by extensions and referred to throughout your application As an escape hatch, they can also be loaded in the query using Ash.Query.load/2, or after the fact using Ash.load/3. Aggregates declared on the resource will be keys in the resource's struct.
See the docs on d:Ash.Resource.Dsl.aggregates for more information.

  
    
  
  Custom aggregates in the query


Custom aggregates can be added to the query and will be placed in the aggregates key of the results. This is an escape hatch, and is not the primary way that you should be using aggregates. It does, however, allow for dynamism, i.e if you are accepting user input that determines what the filter and/or field should be, that kind of thing.
Example:
User
|> Ash.Query.aggregate(
  :count_of_posts,
  :count,
  :posts,
  query: [
    filter: [published: published?]
  ]
)
See the documentation for Ash.Query.aggregate/4 for more information.

  
    
  
  Join Filters


Join filters allows for more complex aggregate queries, including joining with predicates based on multiple related values.

  
    
  
  Example


  aggregates do
    sum :saved_money, [:redeems, :deal], :amount do
      # where any redeem of the deal is redeemed
      filter expr(redeems.redeemed == true)

      # where the `redeems` are `redeemed`
      join_filter :redeems, expr(redeemed == true)

      # where the `redeems.deal.active` == `redeems.require_active`
      join_filter [:redeems, :deal], expr(active == parent(require_active))
    end
  end

  
    
  
  Inline Aggregates


Aggregates can be created in-line in expressions, with their relationship path specified and any options provided that match the options given to Ash.Query.Aggregate.new/4. For example:
calculate :grade, :decimal, expr(
  count(answers, query: [filter: expr(correct == true)]) /
  count(answers, query: [filter: expr(correct == false)])
)
See the Expressions guide for more.



  

    
Validations
    

Validations are similar to changes, except they cannot modify the changeset. They can only continue, or add an error.

  
    
  
  Builtin Validations


There are a number of builtin validations that can be used, and are automatically imported into your resources. See Ash.Resource.Validation.Builtins for more.
Some examples of usage of builtin validations
validate match(:email, ~r/@/)

validate compare(:age, greater_than_or_equal_to: 18) do
  message "must be over 18 to sign up"
end

validate present(:last_name) do
  where [present(:first_name), present(:middle_name)]
  message "must also be supplied if setting first name and middle_name"
end

  
    
  
  Custom Validations


defmodule MyApp.Validations.IsPrime do
  # transform and validate opts

  use Ash.Resource.Validation

  @impl true
  def init(opts) do
    if is_atom(opts[:attribute]) do
      {:ok, opts}
    else
      {:error, "attribute must be an atom!"}
    end
  end

  @impl true
  def validate(changeset, opts, _context) do
    value = Ash.Changeset.get_attribute(changeset, opts[:attribute])
    # this is a function I made up for example
    if is_nil(value) || Math.is_prime?(value) do
      :ok
    else
      # The returned error will be passed into `Ash.Error.to_ash_error/3`
      {:error, field: opts[:attribute], message: "must be prime"}
    end
  end
end
This could then be used in a resource via:
validate {MyApp.Validations.IsPrime, attribute: :foo}

  
    
  
  Anonymous Function Validations


You can also use anonymous functions for validations. Keep in mind, these cannot be made atomic. This is great for prototyping, but we generally recommend using a module, both for organizational purposes, and to allow adding atomic behavior.
validate fn changeset, _context ->
  # put your code here
end

  
    
  
  Where


The where can be used to perform validations conditionally. This functions by running the validation, and if the validation returns an error, we discard the error and skip the operation. This means that even custom validations can be used in conditions.
For example:
validate present(:other_number) do
  where [{MyApp.Validations.IsPrime, attribute: :foo}]
end

  
    
  
  Action vs Global Validations


You can place a validation in any create, update, or destroy action. For example:
actions do
  create :create do
    validate compare(:age, greater_than_or_equal_to: 18)
  end
end
Or you can use the global validations block to validate on all actions of a given type. Where statements can be used in either. Note the warning about running on destroy actions below.
validations do
  validate present([:foo, :bar], at_least: 1) do
    on [:create, :update]
    where present(:baz)
  end
end
The validations section allows you to add validations across multiple actions of a changeset

  
    
  
  Running on destroy actions


By default, validations in the global validations block will run on create and update only. Many validations don't make sense in the context of destroys. To make them run on destroy, use on: [:create, :update, :destroy]


  
    
  
  Examples


validations do
  validate present([:foo, :bar]), on: :update
  validate present([:foo, :bar, :baz], at_least: 2), on: :create
  validate present([:foo, :bar, :baz], at_least: 2), where: [action_is(:action1, :action2)]
  validate absent([:foo, :bar, :baz], exactly: 1), on: [:update, :destroy]
  validate {MyCustomValidation, [foo: :bar]}, on: :create
end

  
    
  
  Atomic Validations


To make a validation atomic, you have to implement the Ash.Resource.Validation.atomic/3 callback. This callback returns an atomic instruction, or a list of atomic instructions, or an error/indication that the validation cannot be done atomically. For our IsPrime example above, this would look something like:
defmodule MyApp.Validations.IsPrime do
  # transform and validate opts

  use Ash.Resource.Validation

  ...

  def atomic(changeset, opts, context) do
    # lets ignore that there is no easy/built-in way to check prime numbers in postgres
    {:atomic,
      # the list of attributes that are involved in the validation
      [opts[:attribute]],
      # the condition that should cause the error
      # here we refer to the new value or the current value
      expr(not(fragment("is_prime(?)", ^atomic_ref(opts[:attribute])))),
      # the error expression
      expr(
        error(^InvalidAttribute, %{
          field: ^opts[:attribute],
          # the value that caused the error
          value: ^atomic_ref(opts[:attribute]),
          # the message to display
          message: ^(context.message || "%{field} must be prime"),
          vars: %{field: ^opts[:attribute]}
        })
      )
    }
  end
end
In some cases, validations operate on arguments only and therefore have no need of atomic behavior. for this, you can call validate/3 directly from atomic/3. The builtin Ash.Resource.Validation.Builtins.argument_equals/2 validation does this, for example.
@impl true
def atomic(changeset, opts, context) do
  validate(changeset, opts, context)
end



  

    
Changes
    

Changes are the primary way of customizing create/update/destroy action behavior. If you are familiar with Plug, you can think of an Ash.Resource.Change as the equivalent of a Plug for changesets. At its most basic, a change will take a changeset and return a new changeset. Changes can be simple, like setting or modifying an attribute value, or more complex, attaching hooks to be executed within the lifecycle of the action.

  
    
  
  Builtin Changes


There are a number of builtin changes that can be used, and are automatically imported into your resources. See Ash.Resource.Change.Builtins for more.
Some examples of usage of builtin changes
# set the `owner` to the current actor
change relate_actor(:owner)

# set `commited_at` to the current timestamp when the action is called
change set_attribute(:committed_at, &DateTime.utc_now/0)

# optimistic lock using the `version` attribute
change optimistic_lock(:version)

  
    
  
  Custom Changes


defmodule MyApp.Changes.Slugify do

  use Ash.Resource.Change

  # transform and validate opts
  @impl true
  def init(opts) do
    if is_atom(opts[:attribute]) do
      {:ok, opts}
    else
      {:error, "attribute must be an atom!"}
    end
  end

  @impl true
  def change(changeset, opts, _context) do
    case Ash.Changeset.fetch_change(changeset, opts[:attribute]) do
      {:ok, new_value} ->
        slug = String.replace(new_value, ~r/\s+/, "-")
        Ash.Changeset.force_change_attribute(changeset, opts[:attribute], slug)
      :error ->
        changeset
    end
  end
end
This could then be used in a resource via:
change {MyApp.Changes.Slugify, attribute: :foo}

  
    
  
  Anonymous Function Changes


You can also use anonymous functions for changes. Keep in mind, these cannot be made atomic, or support batching. This is great for prototyping, but we generally recommend using a module, both for organizational purposes, and to allow adding atomic/batch behavior.
change fn changeset, _context ->
  # put your code here
end

  
    
  
  Where


The where can be used to perform changes conditionally. This functions by running the validations in the where, and if the validation returns an error, we discard the error and skip the operation. This means that even custom validations can be used in conditions.
For example:
change {Slugify, attribute: :foo} do
  where [attribute_equals(:slugify, true)]
end

  
    
  
  Action vs Global Changes


You can place a change on any create, update, or destroy action. For example:
actions do
  create :create do
    change {Slugify, attribute: :name}
  end
end
Or you can use the global changes block to apply to all actions of a given type. Where statements can be used in both cases. Use on to determine the types of actions the validation runs on. By default, it only runs on create and update actions.
changes do
  change {Slugify, attribute: :name} do
    on [:create]
  end
end
The changes section allows you to add changes across multiple actions of a resource.

  
    
  
  Running on destroy actions


By default, changes in the global changes block will run on create and update only. Many changes don't make sense in the context of destroys. To make them run on destroy, use on: [:create, :update, :destroy]


  
    
  
  Examples


changes do
  change relate_actor(:owner)
  change set_attribute(:committed_at, &DateTime.utc_now/0)
  change optimistic_lock(:version), on: [:create, :update, :destroy]
  change {Slugify, [attribute: :foo]}, on: :create
end

  
    
  
  Atomic Changes


To make a change atomic, you have to implement the Ash.Resource.Change.atomic/3 callback. This callback returns a map of changes to attributes that should be changed atomically. We will also honor any Ash.Resource.Change.after_batch/3 functionality to run code after atomic changes have been applied (only if atomic/3 callback has also been defined). Note that Ash.Resource.Change.before_batch/3 is not supported in this scenario and will be ignored.
defmodule MyApp.Changes.Slugify do
  # transform and validate opts

  use Ash.Resource.Change

  ...

  def atomic(changeset, opts, context) do
    {:atomic, %{
      opts[:attribute] => expr(
        fragment("regexp_replace(?, ?, ?)", ^ref(opts[:attribute]), ~r/\s+/, "-")
      )
    }}
  end
end
In some cases, changes operate only on arguments or context, or otherwise can do their work regardless of atomicity. In these cases, you can make your atomic callback call the change/3 function
@impl true
def atomic(changeset, opts, context) do
  {:ok, change(changeset, opts, context)}
end
In other cases, a change may not be necessary in a fully atomic action. For this, you can simply return :ok
@impl true
def atomic(_changeset, _opts, _context) do
  :ok
end

  
    
  
  Batches


Changes can support being run on batches of changesets, using the Ash.Resource.Change.batch_change/3, Ash.Resource.Change.before_batch/3, and Ash.Resource.Change.after_batch/3 callbacks.

  
    
  
  batch_change/3 must be defined


Ash.Resource.Change.before_batch/3 must be defined for Ash.Resource.Change.before_batch/3 and Ash.Resource.Change.after_batch/3 to be called!

For some changes, this may not be necessary at all, i.e the Slugify example has no benefit from batching. If no batch callbacks are added, your change will be run on a loop over the changesets. For the sake of example, however, we will show what it might look like to implement batching for our Slugify example.
defmodule MyApp.Changes.Slugify do
  # transform and validate opts

  use Ash.Resource.Change

  @impl true
  def init(opts) do
    if is_atom(opts[:attribute]) do
      {:ok, opts}
    else
      {:error, "attribute must be an atom!"}
    end
  end

  @impl true
  def batch_change(changeset, opts, context) do
    # here we could run queries or do common work required
    # for a given batch of changesets.
    # in this example, however, we just return the changesets with
    # the change logic applied.
    Enum.map(changesets, &change(&1, opts, context))
  end
end



  

    
Preparations
    

Preparations are the primary way of customizing read action behavior. If you are familiar with Plug, you can think of an Ash.Resource.Preparation as the equivalent of a Plug for queries. At its most basic, a preparation will take a query and return a new query. Queries can be simple, like adding a filter or a sort, or more complex, attaching hooks to be executed within the lifecycle of the action.

  
    
  
  Builtin Preparations


There are builtin preparations that can be used, and are automatically imported into your resources. See Ash.Resource.Preparation.Builtins for more.
The primary preparation you will use is build/1, which passes the arguments through to Ash.Query.build/2 when the preparation is run. See that function for what options can be provided.
Some examples of usage of builtin preparations
# sort by inserted at descending
prepare build(sort: [inserted_at: :desc])

# only show the top 5 results
prepare build(sort: [total_points: :desc], limit: 5)

  
    
  
  Custom Preparations


defmodule MyApp.Preparations.Top5 do
  use Ash.Resource.Preparation

  # transform and validate opts
  @impl true
  def init(opts) do
    if is_atom(opts[:attribute]) do
      {:ok, opts}
    else
      {:error, "attribute must be an atom!"}
    end
  end

  @impl true
  def prepare(query, opts, _context) do
    attribute = opts[:attribute]

    query
    |> Ash.Query.sort([{attribute, :desc}])
    |> Ash.Query.limit(5)
  end
end
This could then be used in a resource via:
prepare {MyApp.Preparations.Top5, attribute: :foo}

  
    
  
  Anonymous Function Queries


You can also use anonymous functions for preparations. This is great for prototyping, but we generally recommend using a module for organizational purposes.
prepare fn query, _context ->
  # put your code here
end

  
    
  
  Action vs Global Preparations


You can place a preparation on a read action, like so:
actions do
  read :read do
    prepare {Top5, attribute: :name}
  end
end
Or you can use the global preparations block to apply to all read actions.
preparations do
  prepare {Top5, attribute: :name}
end
The preparations section allows you to add preparations across multiple actions of a resource.



  

    
Code Interface
    

One of the ways that we interact with our resources is via hand-written code. The general pattern for that looks like building a query or a changeset for a given action, and calling it via functions like Ash.read/2 and Ash.create/2. This, however, is just one way to use Ash, and is designed to help you build tools that work with resources, and to power things like AshPhoenix.Form, AshGraphql.Resource and AshJsonApi.Resource. When working with your resources in code, we generally want something more idiomatic and simple. For example, on a domain called Helpdesk.Support.
resources do
  resource Ticket do
    define :open_ticket, args: [:subject], action: :open
  end
end
This simple setup now allows you to open a ticket with Helpdesk.Support.open_ticket(subject). You can cause it to raise errors instead of return them with Helpdesk.Support.open_ticket!(subject). For information on the options and additional inputs these defined functions take, look at the generated function documentation, which you can do in iex with h Helpdesk.Support.open_ticket. For more information on the code interface, read the DSL documentation: d:Ash.Domain.Dsl.resource.interfaces.

  
    
  
  Code interfaces on the resource


You can define a code interface on individual resources as well, using the code_interface block. The DSL is the same as the DSL for defining it in the domain. For example:
code_interface do
  # the action open can be omitted because it matches the functon name
  define :open, args: [:subject]
end
These will then be called on the resource itself, i.e Helpdesk.Support.Ticket.open(subject).

  
    
  
  Using the code interface


If the action is an update or destroy, it will take a record or a changeset as its first argument.
If the action is a read action, it will take a starting query as an opt in the last argument.
All functions will have an optional last argument that accepts options. See Ash.Resource.Interface.interface_options/1 for valid options.
For reads:
	:query - a query to start the action with, can be used to filter/sort the results of the action.

For creates:
	:changeset - a changeset to start the action with

They will also have an optional second to last argument that is a freeform map to provide action input. It must be a map.
If it is a keyword list, it will be assumed that it is actually options (for convenience).
This allows for the following behaviour:
# Because the 3rd argument is a keyword list, we use it as options
Accounts.register_user(username, password, [tenant: "organization_22"])
# Because the 3rd argument is a map, we use it as action input
Accounts.register_user(username, password, %{key: "val"})
# When all arguments are provided it is unambiguous
Accounts.register_user(username, password, %{key: "val"}, [tenant: "organization_22"])

  
    
  
  Calculations


Resource calculations can be run dynamically using Ash.calculate/3, but
you can also expose them using the code_interface with define_calculation.
For example:
calculations do
  calculate :full_name, :string, expr(first_name <> ^arg(:separator) <> last_name) do
    argument :separator, :string do
      allow_nil? false
      default " "
    end
  end
end

# in your domain
resource User do
  define_calculation :full_name, args: [:first_name, :last_name, {:optional, :separator}]
  # or if you want to take a record as an argument
  define_calculation :full_name, args: [:_record]
end
This could now be used like so:
Accounts.full_name("Jessie", "James", "-")
# or with a record as an argument
Accounts.full_name(user)
This allows for running calculations without an instance of a resource, normally done via Ash.load(user, :full_name)
By default, configured args will be provided for any matching named reference or argument. This is normally fine, but in the case that you have an argument and a reference with the same name, you can specify it by supplying {:arg, :name} and {:ref, :name}. For example:
define_calculation :id_matches, args: [{:arg, :id}, {:ref, :id}]
To make arguments optional, wrap them in {:optional, ..}, for example:
define_calculation :id_matches, args: [{:arg, :id}, {:optional, {:ref, :id}}]

  
    
  
  Bulk & atomic actions



  
    
  
  Bulk Updates & Destroys


Updates support a list, stream, or query as the first argument. This allows for bulk updates. In this mode, an %Ash.BulkResult{} is returned.

  
    
  
  Valid inputs


You cannot provide "any enumerable", only lists, streams (a function or a %Stream{}), and queries. We have to be able to distinguish the input as a bulk input and not input to the action itself.

For example:
Post
|> Ash.Query.filter(author_id == ^author_id)
|> MyApp.Blog.archive_post!()
# => %Ash.BulkResult{}

[%Post{}, %Post{}]
|> MyApp.Blog.destroy_post!()
# => %Ash.BulkResult{}
end
You can pass options to the bulk operation with the bulk_options option to your code interface function.

  
    
  
  Bulk Creates


For bulk creates, you can provide a list or stream of inputs. In this mode also, an %Ash.BulkResult{} is returned.

  
    
  
  Valid inputs


You cannot provide "any enumerable", only lists, streams (a function or a %Stream{}). We have to be able to distinguish the input as a bulk input and not input to the action itself.

Any arguments on the code interface will be applied to all inputs given as a list, and the arguments will come first.
[%{title: "Post 1"}, %{title: "Post 2"}, ...]
# if `:special` is an action argument, it will be applied to all inputs
|> MyApp.Blog.create_post!(:special, bulk_options: [batch_size: 10])

  
    
  
  Returning streams from read actions


The :stream? option allows you to return a stream to be enumerated later.
For example:
MyApp.Blog.my_posts(stream?: true, actor: me)
# => #Stream<...>



  

    
Embedded Resources
    

Embedded resources are stored as maps in attributes of other resources. They are great for storing structured data, and support a whole range of useful features that resources support. For example, you can have calculations, validations, policies and even relationships on embedded resources. Here is an example of a simple embedded resource:
defmodule MyApp.Profile do
  use Ash.Resource,
    data_layer: :embedded # Use the atom `:embedded` as the data layer.

  attributes do
    attribute :first_name, :string, public?: true
    attribute :last_name, :string, public?: true
  end
end

  
    
  
  Embedded resources can't do everything


Embedded resources cannot have aggregates, or expression calculations that rely on data-layer-specific capabilities.


  
    
  
  Adding embedded resource attributes


Embedded resources define an Ash.Type under the hood, meaning you can use them anywhere you would use an Ash type.
defmodule MyApp.User do
  use Ash.Resource, ...

  attributes do
    ...

    attribute :profile, MyApp.Profile, public?: true
    attribute :profiles, {:array, MyApp.Profile}, public?: true # You can also have an array of embeds
  end
end

  
    
  
  Handling nil values


By default, all fields on an embedded resource will be included in the data layer, including keys with nil values. To prevent this, add the embed_nil_values? option to use Ash.Resource. For example:
defmodule YourEmbed do
  use Ash.Resource,
    data_layer: :embedded,
    embed_nil_values?: false
end

  
    
  
  Editing embedded attributes


If you manually supply instances of the embedded structs, the structs you provide are used with no validation. For example:
Ash.Changeset.for_update(user, :update, %{profile: %MyApp.Profile{first_name: "first_name", last_name: "last_name"}})
However, you can also treat embedded resources like regular resources that can be "created", "updated", and "destroyed".
To do this, provide maps as the input, instead of structs. In the example above, if you just wanted to change the first_name, you'd provide:
Ash.Changeset.for_update(user, :update, %{profile: %{first_name: "first_name"}})
This will call the primary update action on the resource. This allows you to define an action on the embed, and add validations to it. For example, you might have something like this:
defmodule MyApp.Profile do
  use Ash.Resource,
    data_layer: :embedded # Use the atom `:embedded` as the data layer.

  attributes do
    attribute :first_name, :string, public?: true
    attribute :last_name, :string, public?: true
  end

  validations do
    validate present([:first_name, :last_name], at_least: 1)
  end
end

  
    
  
  Calculations


Calculations can be added to embedded resources. When you use an embedded resource, you declare what calculations to load via a constraint.
For example:
defmodule MyApp.Profile do
  use Ash.Resource,
    data_layer: :embedded # Use the atom `:embedded` as the data layer.

  attributes do
    attribute :first_name, :string, public?: true
    attribute :last_name, :string, public?: true
  end

  calculations do
    calculate :full_name, :string, concat([:first_name, :last_name], " ")
  end
end

defmodule MyApp.User do
  use Ash.Resource,
    ...

  attributes do
    attribute :profile, MyApp.Profile do
      public? true
      constraints [load: [:full_name]]
    end
  end
end

  
    
  
  Determining what action(s) will be called:


Remember: default actions are already implemented for a resource, with no need to add them. They are called :create, :update, :destroy, and :read. You can use those without defining them. You only need to define them if you wish to override their configuration.

  
    
  
  Single Embeds


	If the current value is nil - a create with the provided values
	If the current value is not nil - an update with the provided values
	If the current value is not nil and the new value is nil - a destroy with the original value


  
    
  
  Array Embeds


All values in the original array are destroyed, and all maps in the new array are used to create new records.

  
    
  
  Adding a primary key


Adding a primary key to your embedded resources is especially useful when managing lists of data. Specifically, it allows you to consider changes to elements with matching primary key values as updates.
For example:
defmodule MyApp.Tag do
  use Ash.Resource,
    data_layer: :embedded

  attributes do
    uuid_primary_key :id
    attribute :name, :string, public?: true
    attribute :counter, :integer, public?: true
  end

  validations do
    validate {Increasing, field: :counter}, on: :update
  end
end
Now, you can accept input meant to update individual list items. The entire list must still be provided, but any items with a matching id will be considered an update instead of a destroy + create.

  
    
  
  Determining what action(s) will be called with a primary key:


Single Embeds with primary keys
	If you provide a struct, instead of a map, the value provided is used as the new relationship value directly.
	If the current value is nil - a create with the provided values
	If the current value is not nil and the primary keys match - an update with the provided values
	If the current value is not nil and the primary keys don't match - a destroy of the original value and a create of the new value
	If the current value is not nil and the new value is nil - a destroy with the original value

Array Embeds with primary keys
	If you provide structs, instead of maps, the value provided is used as the new relationship value directly.
	Any values in the original list with no primary key matching in the new list are destroyed.
	Any values in the new list with no primary key matching in the original list are created.
	Any values with a primary key match in the original list and the new list are updated


  
    
  
  Identities


Identities can be added on an embedded resource, which will ensure that for any given list, they are unique on that identity. For example, if you had an embedded resource called Tag, you could add an identity on name to ensure that nothing has duplicate tag names.

  
    
  
  Usage in Extensions


The AshJsonApi extension exposes these attributes as maps. However, the AshGraphql extension allows you
to specify a type (but not queries/mutations) for an embedded resource. If you do, instead of being treated as a :json type it will get its own named input object type and field type.

  
    
  
  Accessing the source changeset


When building changesets for embedded resources, the source changeset will be available in action changes under changeset.context.__source__.
This allows you to customize the action based on the details of the parent changeset.



  

    
Identities
    

Identities are a way to declare that a record (an instance of a resource) can be uniquely identified by a set of attributes. This information can be used in various ways throughout the framework. The primary key of the resource does not need to be listed as an identity.

  
    
  
  Defining an identity


Identities are defined at the top level of a resource module, eg.
defmodule MyApp.MyResource do
  use Ash.Resource #, ...
  # ...

  identities do
    # If the `email` attribute must be unique across all records
    identity :unique_email, [:email]

    # If the `username` attribute must be unique for every record with a given `site` value
    identity :special_usernames, [:username, :site]
  end
end
See d:Ash.Resource.Dsl.identities for the full range of options available when defining identities.

  
    
  
  Using Ash.get


This will allow these fields to be passed to Ash.get/3, e.g Ash.get(Resource, %{email: "foo"}).

  
    
  
  Using upserts


Create actions support the upsert?: true option, if the data layer supports it. An upsert? involves checking for a conflict on some set of attributes, and translating the behavior to an update in the case one is found. By default, the primary key is used when looking for duplicates, but you can set [upsert?: true, upsert_identity: :identity_name] to tell it to look for conflicts on a specific identity.

  
    
  
  Creating unique constraints


Tools like AshPostgres will create unique constraints in the database automatically for each identity. These unique constraints will honor other configuration on your resource, like the base_filter and attribute multitenancy

  
    
  
  Eager Checking


Setting eager_check?: true on an identity will allow that identity to be checked when building a create changeset over the resource. This allows for showing quick up-front validations about whether some value is taken, for example. If the resource does not have the domain configured, you can specify the domain to use with eager_check_with: DomainName.
If you are using AshPhoenix.Form, for example, this looks for a conflicting record on each call to Form.validate/2.
For updates, it is only checked if one of the involved fields is being changed.
For creates, The identity is checked unless your are performing an upsert, and the upsert_identity is this identity. Keep in mind that for this to work properly, you will need to pass the upsert?: true, upsert_identity: :identity_name when creating the changeset. The primary? read action is used to search for a record. This will error if you have not configured one.

  
    
  
  Pre Checking


pre_check? behaves the same as eager_check?, but it runs just prior to the action being committed. Useful for data layers that don't support transactions/unique constraints, or manual resources with identities. Ash.DataLayer.Ets will require you to set pre_check? since the ETS data layer has no built in support for unique constraints. The domain can be manually specified with pre_check_with: DomainName.



  

    
Notifiers
    


  
    
  
  Built-in Notifiers


Ash comes with a builtin pub_sub notifier: Ash.Notifier.PubSub. See the module documentation for more.

  
    
  
  Creating your own notifier


A notifier is a simple extension that must implement a single callback notify/1. Notifiers do not have to implement an Ash DSL extension, but they may in order to configure how that notifier should behave. See Ash.Notifier.Notification for the currently available fields on a notification.
For more information on creating a DSL extension to configure your notifier, see the docs for Spark.Dsl.Extension.

  
    
  
  Notifier performance


Notifiers should not do intensive synchronous work. If any heavy work needs to be done, they should delegate to something else to handle the notification, like sending it to a GenServer or GenStage.


  
    
  
  Example notifier


defmodule ExampleNotifier do
  use Ash.Notifier

  def notify(%Ash.Notifier.Notification{resource: resource, action: %{type: :create}, actor: actor}) do
    if actor do
      Logger.info("#{actor.id} created a #{resource}")
    else
      Logger.info("A non-logged in user created a #{resource}")
    end
  end
end

  
    
  
  Including a notifier in a resource


If the notifier is also an extension, include it in the notifiers key:
defmodule MyResource do
  use Ash.Resource,
    notifiers: [ExampleNotifier]
end
Configuring a notifier for a specific action or actions can be a great way to avoid complexity in the implementation of a notifier. It allows you to avoid doing things like pattern matching on the action, and treat it more like a change module, that does its work whenever it is called.
create :create do
  notifiers [ExampleNotifier]
end
When your notifier is not an extension, and you want it to run on all actions, include it this way to avoid unnecessary compile time dependencies:
defmodule MyResource do
  use Ash.Resource,
    simple_notifiers: [ExampleNotifier]
end

  
    
  
  Transactions


Domain calls involving resources who's datalayer supports transactions (like Postgres), notifications are saved up and sent after the transaction is closed. For example, the domain call below ultimately results in many many database calls.
Post
|> Ash.Changeset.for_update(:update, %{})
|> Ash.Changeset.manage_relationship(:related_posts, [1, 2, 3], type: :append)
|> Ash.Changeset.manage_relationship(:related_posts, [4, 5], type: :remove)
|> Ash.Changeset.manage_relationship(:comments, [10], type: :append)
|> Ash.update!()
Ash.Changeset.manage_relationship doesn't leverage bulk operations yet, so it performs the following operations:
	a read of the currently related posts
	a read of the currently related comments
	a creation of a post_link to relate to 1
	a creation of a post_link to relate to 2
	a creation of a post_link to relate to 3
	a destruction of the post_link related to 4
	a destruction of the post_link related to 5
	an update to comment 10, to set its post_id to this post

If all three of these resources have notifiers configured, we need to send a notification for each operation (notifications are not sent for reads). For data consistency reasons, if a data layer supports transactions, all writes are done in a transaction. However, if you try to read the record from the database that you have just received a notification about before the transaction has been closed, in a different process, the information will be wrong. For this reason, Ash accumulates notifications until they can be sent.
If you need to perform multiple operations against your resources in your own transaction, you will have to handle that case yourself. To support this, Ash.create/2, Ash.update/2 and Ash.destroy/2 support a return_notifications?: true option. This causes the domain call to return {:ok, result, notifications} in the successful case. Here is an example of how you might use it.
result =
  Ash.DataLayer.transaction(resource, fn ->
    {:ok, something, notifications1} = create_something()
    {:ok, result, notifications2} = create_another_thing(something)
    {:ok, notifications3} = destroy_something(something)

    {result, Enum.concat([notifications1, notifications2, notifications3])}
  end)

case result do
  {:ok, value, notifications} ->
     Ash.Notifier.notify(notifications)

     value
  {:error, error} ->
    handle_error(error)
end



  

    
Actions
    

In Ash, actions are the primary way to interact with your resources. There are five types of actions:
	Read
	Create
	Update
	Destroy
	Generic

All actions can be run in a transaction. Create, update and destroy actions are run in a transaction by default, whereas read and generic actions require opting in with transaction? true in the action definition. Each action has its own set of options, ways of calling it, and ways of customizing it. See the relevant guide for specifics on each action type. This topic focuses on idiomatic ways to use actions, and concepts that cross all action types.

  
    
  
  Primary Actions


Primary actions are a way to inform the framework which actions should be used in certain "automated" circumstances, or in cases where an action has not been specified. If a primary action is attempted to be used but does not exist, you will get an error about it at runtime.
The place you typically need primary actions is when Managing Relationships. When using the defaults option to add default actions, they are marked as primary.
A simple example where a primary action would be used:
# No action is specified, so we look for a primary read.
Ash.get!(Resource, "8ba0ab56-c6e3-4ab0-9c9c-df70e9945281")
To mark an action as primary, add the option, i.e
read :action_name do
  primary? true
end

  
    
  
  Accepting Inputs


Create and Update actions can accept attributes as input. There are two primary ways that you annotate this.

  
    
  
  Using accept in specific actions


Each action can define what it accepts, for example:
create :create do
  accept [:name, :description]
end
You could then pass in %{name: "a name", description: "a description"} to this action.

  
    
  
  Using default_accept for all actions


The resource can have a default_accept, declared in its actions block, which will be used as the accept list for create and destroy actions, if they don't define one.
actions do
  default_accept [:name, :description]

  create :create
  update :update

  update :special_update do
    accept [:something_else]
  end
end
In the example above, you can provide %{name: "a name", description: "a description"} to both the :create and :update actions, but only %{something_else: "some_value"} to :special_update.

  
    
  
  Idiomatic Actions



  
    
  
  Name Your Actions


The intent behind Ash is not to have you building simple CRUD style applications. In a typical set up you may have a resource with four basic actions, there is even a shorthand to accomplish this:
actions do
  defaults [:read, :destroy, create: :*, update: :*]
end
But that is just a simple way to get started, or to create resources that really don't do anything beyond those four operations. You can have as many actions as you want. The best designed Ash applications will have numerous actions, named after the intent behind how they are used. They won't have all reads going through a single read action, and the same goes for the other action types. The richer the actions on the resource, the better interface you can have. With that said, many resources may only have those four basic actions, especially those that are "managed" through some parent resource. See the guide on Managing Relationships for more.

  
    
  
  Put everything inside the action


Ash provides utilities to modify queries and changesets outside of the actions on the resources. This is a very important tool in our tool belt, but it is very easy to abuse. The intent is that as much behavior as possible is put into the action. Here is the "wrong way" to do it. There is a lot going on here, so don't hesitate to check out other relevant guides if you see something you don't understand.
def top_tickets(user_id) do
  Ticket
  |> Ash.Query.for_read(:read)
  |> Ash.Query.filter(priority in [:medium, :high])
  |> Ash.Query.filter(representative_id == ^user_id)
  |> Ash.Query.filter(status == :open)
  |> Ash.Query.sort(opened_at: :desc)
  |> Ash.Query.limit(10)
  |> Helpdesk.Support.read!()
end

# in the resource

actions do 
  defaults [:read, ...]
end
And here is the "right way", where the rules about getting the top tickets have been moved into the resource as a nicely named action, and included in the code_interface of that resource. The reality of the situation is that top_tickets/1 is meant to be obsoleted by your Ash resource! Here is how it should be done.
# in the resource

code_interface do
  define :top, args: [:user_id]
end

actions do 
  read :top do
    argument :user_id, :uuid do
      allow_nil? false
    end

    prepare build(limit: 10, sort: [opened_at: :desc])

    filter expr(priority in [:medium, :high] and representative_id == ^arg(:user_id) and status == :open)
  end
end
Now, whatever code I had that would have called top_tickets/1 can now call Helpdesk.Support.Ticket.top(user.id). By doing it this way, you get the primary benefit of getting a nice simple interface to call into, but you also have a way to modify how the action is invoked in any way necessary, by going back to the old way of building the query manually. For example, if I also only want to see top tickets that were opened in the last 10 minutes:
Ticket
|> Ash.Query.for_read(:top, %{user_id: user.id})
|> Ash.Query.filter(opened_at > ago(10, :minute))
|> Helpdesk.Support.read!()
That is the best of both worlds! These same lessons transfer to changeset based actions as well.



  

    
Read Actions
    

Read actions operate on an Ash.Query. Read actions always return lists of data. The act of pagination, or returning a single result, is handled as part of the interface, and is not a concern of the action itself. Here is an example of a read action:
# Giving your actions informative names is always a good idea
read :ticket_queue do
  # Use arguments to take in values you need to run your read action.
  argument :priorities, {:array, :atom} do
    constraints items: [one_of: [:low, :medium, :high]]
  end

  # This action may be paginated,
  # and returns a total count of records by default
  pagination offset: true, countable: :by_default

  # Arguments can be used in preparations and filters
  filter expr(status == :open and priority in ^arg(:priorities))
end

  
    
  
  Ash.get!


The Ash.get! function is a convenience function for running a read action, filtering by a unique identifier, and expecting only a single result. It is equivalent to the following code:
Ash.get!(Resource, 1)

# is roughly equivalent to

Resource
|> Ash.Query.filter(id == 1)
|> Ash.Query.limit(2)
|> Ash.read!()
|> case do
  [] -> # raise not found error
  [result] -> result
  [_, _] -> # raise too many results error
end

  
    
  
  Ash.read_one!


The Ash.read_one! function is a similar convenience function to Ash.get!, but it does not take a unique identifier. It is useful when you expect an action to return only a single result, and want to enforce that and return a single result.
Ash.read_one!(query)

# is roughly equivalent to

query
|> Ash.Query.limit(2)
|> Ash.read!()
|> case do
  [] -> nil
  [result] -> result
  [_, _] -> # raise too many results error
end

  
    
  
  Pagination


Pagination when reading records is configured on a per-action basis. Ash supports two kinds of pagination: keyset and offset.
A single action can use both kinds of pagination if desired, but typically you would use one or the other.
For pagination configuration reference, see d:Ash.Resource.Dsl.actions.read.pagination.
Counting records
When calling an action that uses pagination, the full count of records can be requested by adding the option page: [count: true].
Note that this will perform a similar query a second time to fetch the count, which can be expensive on large data sets.


  
    
  
  Offset Pagination


Offset pagination is done via providing a limit and an offset when making queries.
	The limit determines how many records should be returned in the query.
	The offset describes how many records from the beginning should be skipped.

Using this, you might make requests like the following:
# Get the first ten records
Ash.read(Resource, page: [limit: 10])
# or by using an action named `read` directly through a
# code interface on the domain
Domain.read(page: [limit: 10])

# Get the next ten records
Ash.read(Resource, page: [limit: 10, offset: 10])
# or by using an action named `read` directly through a
# code interface on the domain
Domain.read(page: [limit: 10, offset: 10])
Next/previous page requests can also be made in memory, using an existing page of search results:
# Return page three of search results
{:ok, third_page} = Resource.read(page: [limit: 10, offset: 20])

# Use `:prev` and `:next` to go backwards and forwards.
# `:first`, `:last`, `:self` and specifying a page number are also supported.
{:ok, second_page} = Ash.page(third_page, :prev)
{:ok, fourth_page} = Ash.page(third_page, :next)
Pros of offset pagination
	Simple to think about
	Possible to skip to a page by number. E.g the 5th page of 10 records is offset: 40
	Easy to reason about what page you are currently on (if the total number of records is requested)
	Can go to the last page (though data may have changed between calculating the last page details, and requesting it)

Cons of offset pagination
	Does not perform well on large datasets (if you have to ask if your dataset is "large", it probably isn't)
	When moving between pages, if data was created or deleted, individual records may be missing or appear on multiple pages


  
    
  
  Keyset Pagination


Keyset pagination is done via providing an after or before option, as well as a limit.
	The limit determines how many records should be returned in the query.
	The after or before value should be a keyset value that has been returned from a previous request. Keyset values are returned whenever there is any read action on a resource that supports keyset pagination, and they are stored in the __metadata__ key of each record.

Keysets are directly tied to the sorting applied to the query
 You can't change the sort applied to a request being paginated, and use the same keyset. If you want to change the sort, but keep the record who's keyset you are using in the before or after option,  you must first request the individual record, with the new sort applied. Then, you can use the new keyset.

For example:
{:ok, page} = Ash.read(Resource, page: [limit: 10])
# Returns `{:ok, %Ash.Page.Keyset{results: [...], before: nil, after: nil}}`
# The `before`/`after` values are the keysets used for this request.

# Fetch the keyset for the next request from the results list
last_record = List.last(page.results)
# Returns `%Resource{__metadata__: %{keyset: "g2wAAAABbQAAACQzOWNjNTcwNy00NjlmL..."}, ...}``

# Use this keyset value to fetch the next page
{:ok, next_page} = Ash.read(Resource, page: [limit: 10, after: last_record.__metadata__.keyset])
Like offset pagination, next/previous page requests can also be made in memory, using an existing page of search results:
# Return page three of search results
{:ok, third_page} = Resource.read(page: [limit: 10])

# Use `:prev` and `:next` to go backwards and forwards.
# `:first` and `:self` can also be used, but `:last` and specifying a page number are not supported.
{:ok, second_page} = Ash.page(third_page, :prev)
{:ok, fourth_page} = Ash.page(third_page, :next)
Pros of keyset pagination
	Performs very well on large datasets (assuming indices exist on the columns being sorted on)
	Behaves well as data changes. The record specified will always be the first or last item in the page

Cons of keyset paginations
	A bit more complex to use
	Can't go to a specific page number


  
    
  
  Example implementation


Setting up the resource
Add the pagination macro call to the action of the resource that you want to be paginated.
defmodule AppName.ResourceName do
  use Ash.Resource

  actions do
    read :read_action_name do
      pagination offset?: true, default_limit: 3, countable: true
    end

    # ...
For all available pagination options, see d:Ash.Resource.Dsl.actions.read|pagination.
Check the updated query return type!
Pagination will modify the return type of calling the query action.
Without pagination, Ash will return a list of records.
But with pagination, Ash will return an Ash.Page.Offset struct (for offset pagination) or Ash.Page.Keyset struct (for keyset pagination). Both structs contain the list of records in the results key of the struct.


  
    
  
  What happens when you call Ash.Query.for_read/4


The following steps are performed when you call Ash.Query.for_read/4.
	Cast input arguments - d:Ash.Resource.Dsl.actions.read.argument
	Set default argument values - d:Ash.Resource.Dsl.actions.read.argument|default
	Add errors for missing required arguments | d:Ash.Resource.Dsl.actions.read.argument|allow_nil?

	Run query preparations | d:Ash.Resource.Dsl.actions.read.prepare

	Add action filter | d:Ash.Resource.Dsl.actions.read|filter



  
    
  
  What happens when you run the action


These steps are trimmed down, and are aimed at helping users understand the general flow. Some steps are omitted.
	Run Ash.Query.for_read/3 if it has not already been run
	Apply tenant filters for attribute
	Apply pagination options
	Run before action hooks
	Multi-datalayer filter is synthesized. We run queries in other data layers to fetch ids and translate related filters to (destination_field in ^ids)
	Strict Check & Filter Authorization is run
	Data layer query is built and validated
	Field policies are added to the query
	Data layer query is Run
	Authorizer "runtime" checks are run (you likely do not have any of these)

The following steps happen while(asynchronously) or after the main data layer query has been run
	If paginating and count was requested, the count is determined at the same time as the query is run.
	Any calculations & aggregates that were able to be run outside of the main query are run
	Relationships, calculations, and aggregates are loaded




  

    
Create Actions
    

Create actions are used to create new records in the data layer. For example:
# on a ticket resource
create :open do
  accept [:title]
  change set_attribute(:status, :open)
end
Here we have a create action called :open that allows setting the title, and sets the status to :open. It could be called like so:
Ticket
|> Ash.Changeset.for_create(:open, %{title: "Need help!"})
|> Ash.create!()
See the Code Interface guide for creating an interface to call the action more elegantly, like so:
Support.open_ticket!("Need help!")

  
    
  
  Bulk creates


Bulk creates take a list or stream of inputs for a given action, and batches calls to the underlying data layer.
Given our example above, you could call Ash.bulk_create like so:
Ash.bulk_create([%{title: "Foo"}, %{title: "Bar"}], Ticket, :open)

  
    
  
  Check the docs!


Make sure to thoroughly read and understand the documentation in Ash.bulk_create/4 before using. Read each option and note the default values. By default, bulk creates don't return records or errors, and don't emit notifications.


  
    
  
  Performance


Generally speaking, all regular Ash create actions are compatible (or can be made to be compatible) with bulk create actions. However, there are some important considerations.
	Ash.Resource.Change modules can be optimized for bulk actions by implementing batch_change/3, before_batch/3 and after_batch/3. If you implement batch_change/3, the change function will no longer be called, and you should swap any behavior implemented with before_action and after_action hooks to logic in the before_batch and after_batch callbacks.

	Actions that reference arguments in changes, i.e change set_attribute(:attr, ^arg(:arg)) will prevent us from using the batch_change/3 behavior. This is usually not a problem, for instance that change is lightweight and would not benefit from being optimized with batch_change/3

	If your action uses after_action hooks, or has after_batch/3 logic defined for any of its changes, then we must ask the data layer to return the records it inserted. Again, this is not generally a problem because we throw away the results of each batch by default. If you are using return_records?: true then you are already requesting all of the results anyway.



  
    
  
  Returning a Stream


Returning a stream allows you to work with a bulk action as an Elixir Stream. For example:
input_stream()
|> Ash.bulk_create(Resource, :action, return_stream?: true, return_records?: true)
|> Stream.map(fn {:ok, result} ->
  # process results
  {:error, error} ->
  # process errors
end)
|> Enum.reduce(%{}, fn {:ok, result}, acc ->
   # process results
   {:error, error} ->
   # process errors
end)

  
    
  
  Be careful with streams


Because streams are lazily evaluated, if you were to do something like this:
[input1, input2, ...] # has 300 things in it
|> Ash.bulk_create(
  Resource,
  :action,
  return_stream?: true,
  return_records?: true,
  batch_size: 100 #  default is 100
)
|> Enum.take(150) # stream has 300, but we only take 150
What would happen is that we would insert 200 records. The stream would end after we process the first two batches of 100. Be sure you aren't using things like Stream.take or Enum.take to limit the amount of things pulled from the stream, unless you actually want to limit the number of records created.


  
    
  
  Upserts


Upserting is the process of "creating or updating" a record, modeled with a single simple create. Both bulk creates and regular creates support upserts. Upserts can be declared in the action, like so:
create :create_user do
  accept [:email]
  upsert? true
  upsert_identity :unique_email
end
Or they can be done with options when calling the create action.
Ash.create!(changeset, upsert?: true, upsert_identity: :unique_email)

  
    
  
  Upserts do not use an update action


While an upsert is conceptually a "create or update" operation, it does not result in an update action being called. The data layer contains the upsert implementation. This means that if you have things like global changes that are only run on update, they will not be run on upserts that result in an update. Additionally, notifications for updates will not be emitted from upserts.


  
    
  
  Atomic Updates


Upserts support atomic updates. These atomic updates do not apply to the data being created. They are only applied in the case of an update. For example:
create :create_game do
  accept [:identifier]
  upsert? true
  upsert_identity :identifier
  change set_attribute(:score, 0)
  change atomic_update(:score, expr(score + 1))
end
This will result in creating a game with a score of 0, and if the game already exists, it will increment the score by 1.
For information on options configured in the action, see d:Ash.Resource.Dsl.actions.create.
For information on options when calling the action, see Ash.create/2.

  
    
  
  What happens when you run a create Action


All actions are run in a transaction if the data layer supports it. You can opt out of this behavior by supplying transaction?: false when creating the action. When an action is being run in a transaction, all steps inside of it are serialized because transactions cannot be split across processes.
	Authorization is performed on the changes
	A before action hook is added to set up belongs_to relationships that are managed. This means potentially creating/modifying the destination of the relationship, and then changing the destination_attribute of the relationship.
	before_transaction and around_transaction hooks are called (Ash.Changeset.before_transaction/2). Keep in mind, any validations that are marked as before_action? true (or all global validations if your action has delay_global_validations? true) will not have happened at this point.
	A transaction is opened if the action is configured for it (by default they are) and the data layer supports transactions
	before_action hooks are performed in order
	The main action is sent to the data layer
	after_action hooks are performed in order
	Non-belongs-to relationships are managed, creating/updating/destroying related records.
	The transaction is closed, if one was opened
	after_transaction hooks are invoked with the result of the transaction (even if it was an error)




  

    
Update Actions
    

Update actions are used to update records in the data layer. For example:
# on a ticket resource
update :close do
  accept [:close_reason]
  change set_attribute(:status, :closed)
end
Here we have an update action called :close that allows setting the close_reason, and sets the status to :closed. It could be called like so:
ticket # providing an initial ticket to close
|> Ash.Changeset.for_update(:close, %{close_reason: "I figured it out."})
|> Ash.update!()
See the Code Interface guide for creating an interface to call the action more elegantly, like so:
Support.close_ticket!(ticket, "I figured it out.")
# You can also provide an id
Support.close_ticket!(ticket.id, "I figured it out.")

  
    
  
  Atomics


Atomic updates can be added to a changeset, which will update the value of an attribute given by an expression. Atomics can be a very powerful way to model updating data in a simple way. An action does not have to be fully atomic in order to leverage atomic updates. For example:
update :add_to_name do
  argument :to_add, :string, allow_nil? false
  change atomic_update(:name, expr("#{name}_#{to_add}"))
end
Changing attributes in this way makes them safer to use in concurrent environments, and is typically more performant than doing it manually in memory.

  
    
  
  Atomics are not stored with other changes


While we recommend using atomics wherever possible, it is important to note that they are stored in their own map in the changeset, i.e changeset.atomics, meaning if you need to do something later in the action with the new value for an attribute, you won't be able to access the new value. This is because atomics are evaluated in the data layer.


  
    
  
  Fully Atomic updates


Atomic updates are a special case of update actions that can be done atomically. If your update action can't be done atomically, you will get an error unless you have set require_atomic? false. This is to encourage you to opt for atomic updates whereever reasonable. Not all actions can reasonably be made atomic, and not all non-atomic actions are problematic for concurrency. The goal is only to make sure that you are aware and have considered the implications.

  
    
  
  What does atomic mean?


An atomic update is one that can be done in a single operation in the data layer. This ensures that there are no issues with concurrent access to the record being updated, and that it is as performant as possible.
For example, the following action cannot be done atomically, because it has
an anonymous function change on it.
update :increment_score do
  change fn changeset, _ ->
    Ash.Changeset.set_attribute(changeset, :score, changeset.data.score + 1)
  end
end
The action shown above is not safe to run concurrently. If two separate processes fetch the record with score 1, and then call increment_score, they will both set the score to 2, when what you almost certainly intended to do was end up with a score of 3
By contrast, the following action can be done atomically
update :increment_score do
  change atomic_update(:score, expr(score + 1)
end
In a SQL data layer, this would produce SQL along the lines of
"UPDATE table SET score = score + 1 WHERE id = post_id"


  
    
  
  What makes an action not atomic?



  
    
  
  Types that can't be atomically casted


Not all types support being casted atomically. For instance, :union types, and embedded resources that have primary keys(and therefore may need to use an update action) cannot currently be casted atomically.

  
    
  
  Changes without an atomic callback


Changes can be enhanced to support atomics by defining Ash.Resource.Change.atomic/3. This callback can return a map of atomic updates to be made to attributes. Here is a simplified example from the built in Ash.Resource.Change.Builtins.increment/2 change:
@impl true
def atomic(_changeset, opts, _context) do
  # Set the requested attribute to its current value (atomic_ref) + the amount
  {:atomic, %{opts[:attribute] => expr(^atomic_ref(opts[:attribute]) + ^opts[:amount])}}
end

  
    
  
  Validations without an atomic callback


Validations can be enhanced to support atomics by defining Ash.Resource.Validation.atomic/3. This callback can return an atomic validation (or a list of atomic validations), which is represented by a list of affected attributes (not currently used), an expression that should trigger an error, and the expression producing the error. Here is an example from the built in Ash.Resource.Validations.Builtins.attribute_equals/2 validation:
@impl true
def atomic(_changeset, opts, context) do
  {:atomic, [opts[:attribute]], expr(^atomic_ref(opts[:attribute]) != ^opts[:value]),
   expr(
     error(^InvalidAttribute, %{
       field: ^opts[:attribute],
       value: ^atomic_ref(opts[:attribute]),
       message: ^(context.message || "must equal %{value}"),
       vars: %{field: ^opts[:attribute], value: ^opts[:value]}
     })
   )}
end

  
    
  
  Bulk updates


There are three strategies for bulk updating data. They are, in order of preference: :atomic, :atomic_batches, and :stream. When calling Ash.bulk_update/4, you can provide a strategy or strategies that can be used, and Ash will choose the best one available. The implementation of the update action and the capabilities of the data layer determine what strategies can be used.

  
    
  
  Atomic


Atomic bulk updates are used when the subject of the bulk update is a query, and the update action can be done atomically and the data layer supports updating a query. They map to a single statement to the data layer to update all matching records. The data layer must support updating a query.

  
    
  
  Example


Ticket
|> Ash.Query.filter(status == :open)
|> Ash.bulk_update!(:close, %{reason: "Closing all open tickets."})
If using a SQL data layer, this would produce a query along the lines of
UPDATE tickets
SET status = 'closed',
    reason = 'Closing all open tickets.'
WHERE status = 'open';

  
    
  
  Atomic Batches


Atomic batches is used when the subject of the bulk update is an enumerable (i.e list or stream) of records and the update action can be done atomically and the data layer supports updating a query. The records are pulled out in batches, and then each batch follows the logic described above. The batch size is controllable by the batch_size option.

  
    
  
  Example



Ash.bulk_update!(one_hundred_tickets, :close, %{reason: "Closing all open tickets."}, batch_size: 10)
If using a SQL data layer, this would produce ten queries along the lines of
UPDATE tickets
SET status = 'closed',
    reason = 'Closing all open tickets.'
WHERE id IN (...ids)

  
    
  
  Stream


Stream is used when the update action cannot be done atomically or if the data layer does not support updating a query. If a query is given, it is run and the records are used as an enumerable of inputs. If an enumerable of inputs is given, each one is updated individually. There is nothing inherently wrong with doing this kind of update, but it will naturally be slower than the other two strategies.
The benefit of having a single interface (Ash.bulk_update/4) is that the caller doesn't need to change based on the performance implications of the action.

  
    
  
  Running a standard update action


All actions are run in a transaction if the data layer supports it. You can opt out of this behavior by supplying transaction?: false when creating the action. When an action is being run in a transaction, all steps inside of it are serialized because transactions cannot be split across processes.
	Authorization is performed on the changes
	A before action hook is added to set up belongs_to relationships that are managed. This means potentially creating/modifying the destination of the relationship, and then changing the destination_attribute of the relationship.
	before_transaction and around_transaction hooks are called (Ash.Changeset.before_transaction/2). Keep in mind, any validations that are marked as before_action? true (or all global validations if your action has delay_global_validations? true) will not have happened at this point.
	A transaction is opened if the action is configured for it (by default they are) and the data layer supports transactions
	before_action hooks are performed in order
	The main action is sent to the data layer
	after_action hooks are performed in order
	Non-belongs-to relationships are managed, creating/updating/destroying related records.
	The transaction is closed, if one was opened
	after_transaction hooks are invoked with the result of the transaction (even if it was an error)


  
    
  
  Atomic Upgrade


Update actions that are run as "normal" update actions will, at the time of execution, be "upgraded" to an atomic action if possible. This  means taking the original inputs and building a corresponding atomic action. This behavior is primarily useful for using things like AshPhoenix.Form, where you want to validate and see the effects of an action before running it, but want the ultimate invocation to be atomic (i.e concurrency safe).
You can disable this by adding atomic_upgrade? false to the action configuration. Additionally, you may want to configure the read action used for atomic upgrades (defaults to the primary read), with atomic_upgrade_with option, i.e atomic_upgrade_with :list_all



  

    
Destroy Actions
    

Destroy actions are comparatively simple. They expect to remove a given record, and by default return :ok in the successful case.
Most destroy actions are one-liners, for example:
destroy :destroy
# Can be added with the defaults
defaults [:read, :destroy]

  
    
  
  Soft Destroy


You can mark a destroy action as soft? true, in which case it is handled by the update action logic.
For example:
destroy :archive do
  soft? true
  change set_attribute(:archived_at, &DateTime.utc_now/0)
end

  
    
  
  Returning the destroyed record


You can use the return_destroyed? option to return the destroyed record.
ticket = Ash.get!(Ticket, 1)
Ash.destroy!(ticket)
# => :ok
ticket = Ash.get!(Ticket, 2)
Ash.destroy!(ticket, return_destroyed?: true)
# => {:ok, %Ticket{}}

  
    
  
  Loading on destroyed records


Keep in mind that using Ash.load on destroyed data will produced mixed results. Relationships may appear as empty, or may be loaded as expected (depending on the data layer/relationship implementation) and calculations/aggregates may show as nil if they must be run in the data layer.


  
    
  
  Bulk Destroys


There are three strategies for bulk destroying data. They are, in order of preference: :atomic, :atomic_batches, and :stream. When calling Ash.bulk_destroy/4, you can provide a strategy or strategies that can be used, and Ash will choose the best one available. The capabilities of the data layer determine what strategies can be used.

  
    
  
  Atomic


Atomic bulk destroys are used when the subject of the bulk destroy is a query and the data layer supports destroying a query. They map to a single statement to the data layer to destroy all matching records.

  
    
  
  Example


Ticket
|> Ash.Query.filter(status == :open)
|> Ash.bulk_destroy!(:close)
If using a SQL data layer, this would produce a query along the lines of
DELETE FROM tickets
WHERE status = 'open';

  
    
  
  Atomic Batches


Atomic batches are used when the subject of the bulk destroy is an enumerable (i.e list or stream) of records and the data layer supports destroying a query. The records are pulled out in batches, and then each batch follows the logic described above. The batch size is controllable by the batch_size option.

  
    
  
  Example



Ash.bulk_destroy!(one_hundred_tickets, :close, %{}, batch_size: 10)
If using a SQL data layer, this would produce ten queries along the lines of
DELETE FROM tickets
WHERE id IN (...ids)

  
    
  
  Stream


Stream is used when the data layer does not support destroying a query. If a query is given, it is run and the records are used as an enumerable of inputs. If an enumerable of inputs is given, each one is destroyed individually. There is nothing inherently wrong with doing this kind of destroy, but it will naturally be slower than the other two strategies.
The benefit of having a single interface (Ash.bulk_destroy/4) is that the caller doesn't need to change based on the performance implications of the action.

  
    
  
  Check the docs!


Make sure to thoroughly read and understand the documentation in Ash.bulk_destroy/4 before using. Read each option and note the default values. By default, bulk destroys don't return records or errors, and don't emit notifications.


  
    
  
  Destroying records


If you provide an enumerable of records, they will be destroyed in batches. For example:
Ash.bulk_destroy([%Ticket{}, %Ticket{}], :destroy)

  
    
  
  Destroying



  
    
  
  Running the Destroy Action


All actions are run in a transaction if the data layer supports it. You can opt out of this behavior by supplying transaction?: false when creating the action. When an action is being run in a transaction, all steps inside of it are serialized because transactions cannot be split across processes.
	Authorization is performed on the changes
	A before action hook is added to set up belongs_to relationships that are managed. This means potentially creating/modifying the destination of the relationship, and then changing the destination_attribute of the relationship.
	before_transaction and around_transaction hooks are called (Ash.Changeset.before_transaction/2). Keep in mind, any validations that are marked as before_action? true (or all global validations if your action has delay_global_validations? true) will not have happened at this point.
	A transaction is opened if the action is configured for it (by default they are) and the data layer supports transactions
	before_action hooks are performed in order
	The main action is sent to the data layer
	after_action hooks are performed in order
	Non-belongs-to relationships are managed, creating/updating/destroying related records.
	The transaction is closed, if one was opened
	after_transaction hooks are invoked with the result of the transaction (even if it was an error)




  

    
Generic Actions
    

Generic actions are so named because there are no special rules about how they work. A generic action takes arguments and returns a value. The struct used for building input for a generic action is Ash.ActionInput.
action :say_hello, :string do
  argument :name, :string, allow_nil?: false

  run fn input, _ ->
    {:ok, "Hello: #{input.arguments.name}"}
  end
end
A generic action declares its arguments, return type, and implementation, as illustrated above.

  
    
  
  No return? No problem!


Generic actions can omit a return type, in which case running them returns :ok if successful.
action :schedule_job do
  argument :job_name, :string, allow_nil?: false
  run fn input ->
    # Schedule the job
    :ok
  end
end


  
    
  
  Why use generic actions?


The example above could be written as a normal function in elixir, i.e
def say_hello(name), do: "Hello: #{name}"
The benefit of using generic actions instead of defining normal functions:
	They can be used with api extensions like ash_json_api and ash_graphql
	Their inputs are type checked and casted
	They support Ash authorization patterns (i.e policies)
	They can be included in the code interface of a resource
	They can be made transactional with a single option (transaction? true)

If you don't need any of the above, then there is no problem with writing regular Elixir functions!

  
    
  
  Return types and constraints


Generic actions do not cast their return types. It is expected that the action return a valid value for the type that they declare. However, declaring additional constraints can inform API usage, and make the action more clear. For example:
action :priority, :integer do
  constraints [min: 1, max: 3]
  argument :status, :atom, constraints: [one_of: [:high, :medium, :low]]

  run fn input, _ ->
    case input.arguments.status do
      :high -> {:ok, 3}
      :medium -> {:ok, 2}
      :low -> {:ok, 1}
    end
  end
end
Returning resource instances
It sometimes happens that you want to make a generic action which returns an
instance or instances of the resource. It's natural to assume that you can
set your action's return type to the name of your resource. This won't work
as resources do not define a type, unless they are embedded. In embedded resources, this won't work because the module is still being compiled, so referencing yourself as a type causes a compile error. Instead, use the :struct type and the instance_of constraint, like so:
action :get, :struct do
  constraints instance_of: __MODULE__

  run # ...
end




  

    
Manual Actions
    

Manual actions allow you to control how an action is performed instead of dispatching to a data layer. To do this, specify the manual option with a module that adopts the appropriate behavior.
Manual actions are a way to implement an action in a fully custom way. This can be a very useful escape hatch when you have something that you are finding difficult to model with Ash's builtin tools.

  
    
  
  Manual Creates/Updates/Destroy


For manual create, update and destroy actions, a module is passed that uses one of the following (Ash.Resource.ManualCreate, Ash.Resource.ManualUpdate and Ash.Resource.ManualDestroy).
For example:
create :special_create do
  manual MyApp.DoCreate
end

# The implementation
defmodule MyApp.DoCreate do
  use Ash.Resource.ManualCreate

  def create(changeset, _, _) do
    record = create_the_record(changeset)
    {:ok, record}

    # An `{:error, error}` tuple should be returned if something failed
  end
end
The underlying record can be retrieved from changeset.data for update and destroy manual actions. The changeset given to the manual action will be after any before_action hooks, and before any after_action hooks.

  
    
  
  Manual Read Actions


Manual read actions work the same, except the will also get the "data layer query". For AshPostgres, this means you get the ecto query that would have been run. You can use Ash.Query.apply_to/3 to apply a query to records in memory. This allows you to fetch the data in a way that is not possible with the data layer, but still honor the query that was provided to.
# in the resource
actions do
  read :action_name do
    manual MyApp.ManualRead
    # or `{MyApp.ManualRead, ...opts}`
  end
end

# the implementation
defmodule MyApp.ManualRead do
  use Ash.Resource.ManualRead

  def read(ash_query, ecto_query, _opts, _context) do
    ...
    {:ok, query_results} | {:error, error}
  end
end

  
    
  
  Modifying the query


As an alternative to manual read actions, you can also provide the modify_query option, which takes an MFA and allows low level manipulation of the query just before it is dispatched to the data layer.
For example:
read :read do
  modify_query {MyApp.ModifyQuery, :modify, []}
end

defmodule MyApp.ModifyQuery do
  def modify(ash_query, data_layer_query) do
    {:ok, modify_data_layer_query(data_layer_query)}
  end
end
This can be used as a last-resort escape hatch when you want to still use resource actions but need to do something that you can't do easily with Ash tools. As with any low level escape hatch, here be dragons.

